HEWLETT PACKARD

HP-71

Users' Library Solutions

Math

HEWLETT PACKARD

USERS' LIBRARY ORDER FORM

HP-71 Solutions Books are available at a cost of $12.50 each. To order, please call toll free (800) 538-8787, contact your local dealer, or use the form provided below to order directly from the Users' Library. If ordering from the Users' Library, please include a handling charge of $3.50 per order.

REMEMBER...

You can further customize your HP-71 with software available through the Users' Library.

A Users' Library Membership includes 2 free Solutions Books (a $25.00 value) plus periodic updates containing:

- –Product Information
- –Contests
- –Discounts
- –New Software Availability
- –Third-Party Products

YES! I want to subscribe to the Users' Library.

☐ $20.00 U.S. and Canada (5851-1011)
(plus state and local taxes)

☐ $35.00 OUTSIDE the U.S. and Canada (5851-1012)

Solution Books Requested

1. _____
Title Part Number

2. _____
Title Part Number

HP-71 Solutions Book Selection includes:

00071-90064/Math
00071-90065/Games
00071-90066/General Utilities

Please contact the Users' Library for further information regarding recorded material.

Solutions Book Request

Solutions Book Part Number	Title	Price
00071-900 _ _		
00071-900 _ _		
00071-900 _ _		
	Users' Library Subscription	
	State and Local Taxes	
	Solutions Book handling charge ($3.50)	
	TOTAL	

All payments must be in U.S. dollars and drawn on a U.S. bank.

Send all orders (with payment) directly to: HEWLETT-PACKARD USERS' LIBRARY
Dept. 39UL
1000 N.E. Circle Blvd.
Corvallis, Oregon 97330

☐ Payment enclosed ☐ Master Card ☐ VISA ☐ Purchase order
($20.00 minimum)

_____
Card Number/Company Purchase Order

_____
Name

_____
Company (if applicable)

_____
Address

_____
City State/Country Zip

CONTENTS

Vector Operations

This program provides solutions to the most common vector operations, such as addition, subtraction, dot and cross products, included angle, multiplication of a vector by a scalar, finding the length of a vector, and determining unit vectors. The program will allow vectors to be entered in either rectangular (x, y, z) or cylindrical (r, theta, z) coordinates, and will also display the result in either form. Conversion between these two formats is also an option. A useful feature of this program is the ability to chain vector operations by using the result from one operation as an operand for the next one.

Let V1 = a1X + b1Y + c1Z and V2= a2X + b2Y + c2Z

V1 + V2 = (a1+a2)X + (b1+b2)Y + (c1+c2)Z

V1 - V2 = (a1-a2)X + (b1-b2)Y + (c1-c2)Z

V1 X V2 = (b1*c2-b2*c1)X + (c1*a2-c1*c2)Y + (a1*b2-a2*b1)Z

V1 . V2 = (a1*a2)X + (b1*b2)Y + (c1*c2)Z

V1*(s) = (a1*s)X + (b1*s)Y + (c1*s)Z

norm(V1) = (a1*a1 + b1*b1 + c1*c1)^.5

unit vector(V1) = (a1/norm(V1))X + (b1/norm(v1))Y + (c1/norm(v1))Z

included angle between V1 & V2 =

$$\text{acos}\left[\frac{(\text{norm}(V1)^2 + \text{norm}(V2)^2 - \text{norm}(V3)^2)}{2*\text{norm}(V1)*\text{norm}(V2)}\right]$$

V3 = vector between points defining V1 and V2

Comments	Input	Display
1) Run the program.		INPUT R/C ?
2) The user may enter vectors in either rectangular or cylindrical form. Press [R] to enter x, y, z components of the vector or [C] to enter r, theta and z components of the vector. If [C] is chosen, user flag 1 will be set as a visual reminder of the input format required.	[R] or [C]	DISPLAY R/C ?
3) If the result of a vector operation is a vector, the program will display the result in either rectangular or cylindrical form. Press [R] to see the x, y, and z components of the result, or press [C] to see the result expressed in terms of r, theta and z. If [C] is chosen, user flag 2 will be set as a visual reminder of the display format.	[R] or [C]	A,S,X,I,D,N,M,U,C,F,Q?
4) The program is prompting for the vector operation to be performed. The options are described below and grouped according to their input requirements. V1 represents the first vector entered, and V2 the second (if necessary).		

****
* SINGLE VECTOR OPERATIONS *
****

N: calculate the norm or magnitude of a vector and return the scalar result. [N]

M: multiply a vector by a scalar and return the vector result in the proper format. [M]

Comments	Input	Display

U: calculate the unit vector with the same direction as the input vector and with a norm of 1. Return the vector result in the proper format. [U]

C: convert the input vector from rectangular to cylindrical or vice versa. This option requires that the input and display formats chosen by the user correspond to the desired conversion. If they do not, the message 'I/O INCORRECT' will appear and the program will again ask for input and display formats. [C]

For these single vector operations, one of the following two sets of input prompts will appear, depending on the input mode selected.

		x=
* rectangular *	<x>[ENDLINE]	y=
* coordinates *	<y>[ENDLINE]	z=
	<z>[ENDLINE]	
		r=
* cylindrical *	<r>[ENDLINE]	theta=
* coordinates *	<theta>[ENDLN]	z=
	<z>[ENDLINE]	

```
*******************************
*    TWO VECTOR OPERATIONS    *
*******************************
```

A: add V1 and V2 and return the result in the proper format. [A]

Comments	Input	Display
S: subtract V2 from V1 and return the result in the proper format.	[S]	
X: calculate the cross product V1 x V2 and return the result in the proper format.	[X]	
I: calculate the included angle between V1 and V2 and return the result in degrees.	[I]	
D: calculate the dot or scalar product V1.V2 and return the scalar result.	[D]	
For these two-vector operations, one of the following two sets of input prompts will appear, depending on the input mode selected.		
* rectangular * coordinates *	<x1>[ENDLINE] <y1>[ENDLINE] <z1>[ENDLINE] <x2>[ENDLINE] <y2>[ENDLINE] <z2>[ENDLINE]	VECTOR 1: x= VECTOR 1: y= VECTOR 1: z= VECTOR 2: x= VECTOR 2: y= VECTOR 2: z=
* cylindrical * coordinates *	<r1>[ENDLINE] <theta1>[ENDLN] <z1>[ENDLINE] <r2>[ENDLINE] <theta2>[ENDLN] <z2>[ENDLINE]	VECTOR 1: r= VECTOR 1: theta= VECTOR 1: z= VECTOR 2: r= VECTOR 2: theta= VECTOR 2: z=

****
* OTHER OPTIONS *
****

F: allows the user to change input and/or display formats from cylindrical to rectangular or vice versa.	[F]	<go to step 2>

Comments	Input	Display
Q: exit program.	[Q]	>

5) After an operation is chosen, the program will return the result in one of the formats listed below, depending on whether it is a vector or scalar. Pressing any key (except [ON]) will display the next component or move on to step 6.

****
* VECTOR RESULTS *
****

[A],[S],[X],[M],[U],[C] : These options will return a vector result in either rectangular or cylindrical coordinates, depending on the display option chosen in step 3.

		x=
* rectangular *	<any key>	y=
* coordinates *	<any key>	z=
	<any key>	<go to step 6>
		r=
* cylindrical *	<any key>	theta=
* coordinates *	<any key>	z=
	<any key>	<go to step 6>

****
* SCALAR RESULTS *
****

[I]	INCLUDED ANGLE =
<any key>	<go to step 6>
[D]	DOT PRODUCT =
	<go to step 6>
[N]	NORM =
	<go to step 6>

Comments	Input	Display
6) At this point, the user is given the option of exiting the program or running it again.		RUN AGAIN (Y/N)?
To exit program, press 'N'.	[N]	>
To run again, press 'Y'.	[Y]	<go to step 7>
7) If the result of the last operation was a scalar, the program will return to step 4.		
If it was a vector, the program provides the option of using it in subsequent calculations.		USE RESULT (Y/N)?
To discard the result, press 'N'. To use the result, press 'Y'.	[N] or [Y]	<go to step 4>
If 'Y' is chosen and the next operation requires one vector, the program will skip the prompt for the components and display the result. If two vectors are required, the program will assume the result is the first vector and will ask only for the second vector.		

REFERENCES:

Salas, S. and Hille, E.; "Calculus", Xerox.

Hudson, R.; "The Engineer's Manual", Wiley and Sons, Inc.

EXAMPLES

A) Find the cross product of (3,5,Ø) and (4,Ø,1), and then calculate the angle the result vector makes with the x-axis (1,Ø,Ø).

B) Find the length (norm) of the vector (12.66,-4.5,-7).

C) Convert (-4,7,Ø) to cylindrical coordinates.

Comments	Input	Display
1) Run the program.	[RUN]	INPUT R/C ?
2) Specify input in rectangular coordinates.	[R]	DISPLAY R/C ?
3) Specify output in rectangular coordinates.	[R]	A,S,X,I,D,N,M,U,C,F,Q?
4) Select cross product option.	[X]	VECTOR 1: x=
5) Enter x, y, z for vectors.	3[ENDLINE]	VECTOR 1: y=
	5[ENDLINE]	VECTOR 1: z=
	Ø[ENDLINE]	VECTOR 2: x=
	4[ENDLINE]	VECTOR 2: y=
	Ø[ENDLINE]	VECTOR 2: z=
	1[ENDLINE]	
6) Display result.		x=5
	<any key>	y=-3
	<any key>	z=-2Ø
7) Run program again.	<any key>	RUN AGAIN (Y/N)?
8) Use result.	[Y]	USE RESULT (Y/N)?
9) Calculate included angle.	[Y]	A,S,X,I,D,N,M,U,C,F,Q?
1Ø) Enter second vector.	[I]	VECTOR 2: x=
	1[ENDLINE]	VECTOR 2: y=
	Ø[ENDLINE]	VECTOR 2: z=
	Ø[ENDLINE]	
11) Display result.		ANGLE = 76.1130063355
12) Run program again.	<any key>	RUN AGAIN? (Y/N)
13) Calculate norm.	[Y]	A,S,X,I,D,N,M,U,C,F,Q?
14) Input vector.	[N]	x=
	12.66[ENDLINE]	y=
	-4.5 [ENDLINE]	z=
	-7 [ENDLINE]	
15) Display result.		NORM = 15.1501023099

EXAMPLES

Comments	Input	Display
16) Run program again.	<any key>	RUN AGAIN? (Y/N)
17) Convert from rectangular to cylindrical.	[Y]	A,S,X,I,D,N,M,U,C,F,Q?
18) Change output format to match conversion.	[C]	I/O INCORRECT INPUT R/C?
19) Specify rectangular input.	[R]	DISPLAY R/C?
20) Specify cylindrical output.	[C]	A,S,X,I,N,M,U,C,F,Q?
21) Choose conversion now that I/O format is correct.	[C]	x=
22) Input vector.	-4[ENDLINE] 7 [ENDLINE] 0 [ENDLINE]	y= z=
23) Display result.	<any key> <any key>	r= 8.0622577483 theta= 119.744881297 z= 0
24) End program.	<any key> [N]	RUN AGAIN? (Y/N) >

```
ØØ1Ø ! VECTOR OPERATIONS
ØØ2Ø ! Revision 1.Ø  3/28/84
ØØ3Ø !
ØØ4Ø DEGREES @ DELAY Ø,Ø @ CFLAG 1,2,3,4
ØØ5Ø KØ$="ASXIDNMUCFQ" @ IØ$="CR" @ Il$="YN"
ØØ6Ø 'IO':
ØØ7Ø DISP "INPUT R/C ? ";
ØØ8Ø 'Il': Kl$=UPRC$(KEY$) @ IF NOT POS(IØ$,Kl$) THEN 'Il' ELSE DISP
ØØ9Ø IF Kl$="C" THEN SFLAG 1 ELSE CFLAG 1
Ø1ØØ IF FLAG(1) THEN Xl$="r=" @ Yl$="theta=" ELSE Xl$="x=" @ Yl$="y="
Ø11Ø DISP "DISPLAY R/C ? ";
Ø12Ø 'Ol': Kl$=UPRC$(KEY$) @ IF NOT POS(IØ$,Kl$) THEN 'Ol' ELSE DISP
Ø13Ø IF Kl$="C" THEN SFLAG 2 ELSE CFLAG 2
Ø14Ø IF FLAG(2) THEN X2$="r=" @ Y2$="theta=" ELSE X2$="x=" @ Y2$="y="
Ø15Ø 'OP': DISP "A,S,X,I,D,N,M,U,C,F,Q"
Ø16Ø 'WAIT': Kl$=UPRC$(KEY$) @ KØ=POS(KØ$,Kl$) @ IF NOT KØ THEN 'WAIT'
Ø17Ø IF KØ=11 THEN 'Q'
Ø18Ø IF KØ=1Ø THEN CFLAG 1,2,3 @ GOTO 'IO'
Ø19Ø IF KØ#9 THEN 'Al'
Ø2ØØ IF NOT FLAG(1) EXOR FLAG(2) THEN DISP 'I/O INCORRECT' @ WAIT 2 ELSE 'Al'
Ø21Ø GOTO 'IO'
Ø22Ø 'Al': IF KØ<6 THEN 'I2VA'
Ø23Ø 'IlV': IF FLAG(3) THEN 'SUBCALL'
Ø24Ø DISP Xl$; @ INPUT "";Xl
Ø25Ø IF FLAG(1) AND Xl<Ø THEN DISP "INVALID ENTRY" @ GOTO 'IlV'
Ø26Ø DISP Yl$; @ INPUT "";Yl
Ø27Ø INPUT "z=";Zl
Ø28Ø GOTO 'SUBCALL'
Ø29Ø 'I2VA': IF FLAG(3) THEN 'I2VB'
Ø3ØØ DISP "VECTOR 1:"&Xl$; @ INPUT "";Xl
Ø31Ø IF FLAG(1) AND Xl<Ø THEN DISP "INVALID ENTRY" @ GOTO 'I2VA'
Ø32Ø DISP "VECTOR 1:"&Yl$; @ INPUT "";Yl
Ø33Ø INPUT "VECTOR 1:z=";Zl
Ø34Ø 'I2VB':
Ø35Ø DISP "VECTOR 2:"&Xl$; @ INPUT "";X2
Ø36Ø IF FLAG(1) AND X2<Ø THEN DISP "INVALID ENTRY" @ GOTO 'I2VB'
Ø37Ø DISP "VECTOR 2:"&Yl$; @ INPUT "";Y2
Ø38Ø INPUT "VECTOR 2:z=";Z2
Ø39Ø IF FLAG(1) THEN CALL C2R(X2,Y2)
Ø4ØØ 'SUBCALL':
Ø41Ø IF FLAG(1) THEN CALL C2R(Xl,Yl)
Ø42Ø GOSUB Kl$
Ø43Ø 'CYCLE':
Ø44Ø DISP "RUN AGAIN? (Y/N) ";
Ø45Ø 'A2': Kl$=UPRC$(KEY$) @ ON POS(Il$,Kl$)+1 GOTO 'A2','USEIT','Q'
Ø46Ø 'USEIT': DISP @ IF IP((KØ-1)/3)=1 OR KØ=9 THEN CFLAG 3 @ GOTO 'OP'
Ø47Ø DISP "USE RESULT? (Y/N)";
Ø48Ø 'A3': Kl$=UPRC$(KEY$) @ IF NOT POS(Il$,Kl$) THEN 'A3' ELSE DISP
Ø49Ø IF POS("N",Kl$) THEN CFLAG 3 @ GOTO 'OP'
Ø5ØØ IF NOT POS("Y",Kl$) THEN 'USEIT'
Ø51Ø IF FLAG(2) THEN CALL C2R(XØ,YØ)
Ø52Ø Xl=XØ @ Yl=YØ @ Zl=ZØ @ SFLAG 3 @ GOTO 'OP'
Ø53Ø 'A':
Ø54Ø XØ=Xl+X2 @ YØ=Yl+Y2 @ ZØ=Zl+Z2
Ø55Ø GOSUB 'OV'
```

```
0560 RETURN
0570 'S':
0580 X0=X1-X2 @ Y0=Y1-Y2 @ Z0=Z1-Z2
0590 GOSUB 'OV'
0600 RETURN
0610 'D':
0620 D0=X1*X2+Y1*Y2+Z1*Z2
0630 DISP "DOT PRODUCT =";D0 @ GOSUB 'W'
0640 RETURN
0650 'X':
0660 X0=Y1*Z2-Y2*Z1 @ Y0=Z1*X2-Z2*X1 @ Z0=X1*Y2-X2*Y1
0670 GOSUB 'OV'
0680 RETURN
0690 'M': INPUT "SCALAR =";S
0700 X0=X1*S @ Y0=Y1*S @ Z0=Z1*S
0710 GOSUB 'OV'
0720 RETURN
0730 'I':
0740 SFLAG 4 @ GOSUB 'N' @ CFLAG 4
0750 M3=SQR((X2-X1)^2+(Y2-Y1)^2+(Z2-Z1)^2)
0760 A0=(M1*M1+M2*M2-M3*M3)/(2*M1*M2)
0770 C9=.000000000002
0780 IF FP(FP(ABS(A0))+C9)<.000000000045 THEN A0=IP(A0+SGN(A0)*C9)
0790 A1=ACOS(A0)
0800 DISP "INCLUDED ANGLE =";A1 @ GOSUB 'W'
0810 RETURN
0820 'N':
0830 M1=SQR(X1*X1+Y1*Y1+Z1*Z1)
0840 M2=SQR(X2*X2+Y2*Y2+Z2*Z2)
0850 IF NOT FLAG(4) THEN DISP "NORM =";M1 @ GOSUB 'W'
0860 RETURN
0870 'U':
0880 SFLAG 4 @ GOSUB 'N' @ CFLAG 4
0890 X0=X1/M1 @ Y0=Y1/M1 @ Z0=Z1/M1
0900 GOSUB 'OV'
0910 RETURN
0920 'W': IF KEY$="" THEN 'W' ELSE RETURN
0930 'C':
0940 X0=X1 @ Y0=Y1 @ Z0=Z1
0950 'OV':
0960 IF FLAG(2) THEN CALL R2C(X0,Y0)
0970 DISP X2$;X0 @ GOSUB 'W'
0980 DISP Y2$;Y0 @ GOSUB 'W'
0990 DISP "z=";Z0 @ GOSUB 'W'
1000 RETURN
1010 'Q': CFLAG 1,2,3,4 @ DISP @ PUT "#43" @ END
1020 SUB R2C(R,T)
1030 I=R @ J=T
1040 R=SQR(I*I+J*J)
1050 T=ANGLE(I,J)
1060 SUB END
1070 SUB C2R(I,J)
1080 R=I @ T=J
1090 I=R*COS(T)
1100 J=R*SIN(T)
1110 SUB END
```

Numerical Integration

This program will perform numerical integration whether a function is known explicitly or at a finite number of equally spaced points.

For the explicit case, a 16 point Gaussian quadrature is provided for a finite interval. Also, the integral may be found for the explicit case using Simpson's or the Trapezoidal methods. The quadrature is more accurate, however.

Given a finite set of points, Simpson's rule or the Trapezoidal rule may be used to find the integral.

Equations:

Gaussian Quadrature:

$$(b-a)/\ 2 \sum_{j=1}^{N} w<j>f((a+b+x<j>(b-a))/2)$$

where w<j>,x<j> are the weights and nodes, respectively. The weights and nodes are for an integral from -1 to 1.

Simpson's Rule:

h[f(xØ) + 4f(x1) + 2f(x2) + .. + 2f(x<n-2>) + 4f(x<n-1>) + f(x<n>)]/3

where h is the (equally spaced) interval distance.
'n' must be even for the method to work.

Trapezoidal Rule:

$$h/2[f(xØ) +2 \sum_{j=1}^{n=1} (f(x<j>) + f(x<n>)]$$

where h = (b-a)/n for interval (a,b).

Comments	Input	Display
1) Run the program.		Gauss,Simp,Trap,Quit
2) Choose desired method. a. Gauss b. Simpson c. Trapezoidal d. Quit Go to desired option to continue.	 G S T Q	 f(x)= Explicit (Y/N) Explicit (Y/N) <blinking cursor>
****** Gauss ******		
G1) Key in the function with variable 'x'.	<function> [ENDLINE]	 Lwr,Upr bounds=
G2) Key in the lower and upper bound separated by a comma.	<L,U>	calculating.. Result= <val.>
G3) To continue, press any key. Go to step 1 to continue.	<any key>	Gauss,Simp,Trap,Quit
****** Simpson ******		
S1) If the function is known: Go to the explicit case (S5). If the function is unknown:	 Y N	 # of Partitons=<val.> Nmbr of pnts=<val.>
S2) (Implicit case) Key in the number of points. For Simpson's rule, this must be an odd number.	<number>	Interval length=<val.>
S3) Key in the interval length for the partitions. This routine requires that partitions be the same length.	<length>	E(Ø)=<val.>
You are now in the matrix editor. Continue at that section (E1) and return.		calculating.. Result=<val.>
S4) To continue, press any key. Go to step 1 to continue.	<any key>	Gauss,Simp,Trap,Quit
S5) (Explicit Case) Key in the number of partitions desired. This must be an even value.	<number>	f(x)=

Comments	Input	Display
S6) Key in the function with variable 'x'.	<function> [ENDLINE]	calculating.. Result= <val.>
S7) Key in the lower and upper bound separated by a comma.	<lb,ub>	calculating.. Result= <val.>
S8) To continue, press any key. Go to step 1 to continue.	<any key>	Gauss,Simp,Trap,Quit
****** Trapezoidal ******		
T1) If the function is known: Go to the explicit case (T5). If the function is unknown:	Y N	# of Partitons=<val.> Nmbr of pnts=<val.>
T2) (Implicit case) Key in the number of points.	<number>	Interval length=<val.>
T3) Key in the interval length for the partitions.	<length>	E(Ø)=<val.>
You are now in the matrix editor. Continue at that section (E1) and return.		calculating.. Result=<val.>
T4) To continue, press any key. Go to step 1 to continue.	<any key>	Gauss,Simp,Trap,Quit
T5) (Explicit Case) Key in the number of partitions desired.	<number>	f(x)=
T6) Key in the function with variable 'x'.	<function> [ENDLINE]	calculating.. Result= <val.>
T7) Key in the lower and upper bound separated by a comma.	<lb,ub>	calculating.. Result= <val.>
T8) To continue, press any key. Go to step 1 to continue.	<any key>	Gauss,Simp,Trap,Quit

Comments	Input	Display
****** Matrix Editor ******		
The points E(Ø) through E(n) may be entered. The value displayed is the current value of the point. You may change the value, move to a previous value, move to the next value, or specify a point to move to. Pressing [Q] exits the matrix editor, and the program finds the value of the integral.		E(I)=<value>
E1) To change value:	[ENDLINE] <new val.> [ENDLINE]	E(I+1)=<value>
To move to next value (this is the arrow key, not the greater than char.):	[>]	E(I+1)=<value>
To move to previous value (this is the arrow key, not the less than char.):	[<]	E(I-1)=<value>
To move to a specific point: Key in the desired element index:	[SPC] <index>	Element= E(<index>)=<val.>
To quit the routine:	[Q]	calculating..

EXAMPLE PROBLEMS

Problem 1.

Given the approximations below for f(x), compute the approximations to the integral from the bounds Ø-2 by the trapezoidal rule and by Simpson's rule. The interval length is Ø.25.

i	Ø	1	2	3	4	5	6	7	8
f(x)	2	2.8	3.8	5.2	7	9.2	12.1	15.6	2Ø

Comments	Input	Display
1) Run the program.		Gauss,Simp,Trap,Quit
2) Choose trapezoidal method:	T	Explicit (Y/N)
3) Since the function is unknown, choose implicit case:	N	Nmbr of pnts=Ø
4) Key in the number of points.	9 [ENDLINE]	Interval length=Ø
5) Key in the interval length for the partitions.	.25 [ENDLINE]	E(Ø)=Ø
6) Enter the function values:	[ENDLINE]	value=
	2 [ENDLINE]	E(1)=Ø
	[ENDLINE]	value=
	2.9 [ENDLINE]	E(2)=Ø
The previous value was supposed to have been 2.8. Edit it.	[<]	E(1)=2.9
	[ENDLINE]	value =
	2.8 [ENDLINE]	E(2)=Ø
Continue entering the rest of of the values in the same manner.		
To quit:	Q	calculating.. Result=16.6750
7) To continue, press any key.	<any key>	Gauss,Simp,Trap,Quit
8) Now use Simpson's rule.	S	Explicit (Y/N)
9) Since the function is unknown, choose implicit case:	N	Nmbr of pnts=9

Comments	Input	Display
10) Since this is the correct number, just press:	[ENDLINE]	Interval length=.2500
11) This value is also okay.	[ENDLINE]	E(0)=2
12) Since the program leaves the values the same, simply press [Q] to finish.	Q	calculating.. Result=16.5833
13) To continue, press any key.	<any key>	Gauss,Simp,Trap,Quit

Problem 2.

Find the value of the integral of f(x) from 0 to 2*pi where f(x) = 1/(1-cos(x)+ .25).

Comments	Input	Display
1) Run the program.		Gauss,Simp,Trap,Quit
2) Choose Gauss's method:	G	f(x)=
3) Key in the function with variable 'x'.	1/(1-cos(x)+.25) [ENDLINE]	Lwr,Upr bounds=
4) Key in the lower and upper bound separated by a comma.	0,2*pi [ENDLINE]	calculating.. result= 8.3776
5) To continue, press any key.	<any key>	Gauss,Simp,Trap,Quit

Note that some of the comments are not preceded by line numbers.

```
ØØ1Ø DEF FNK$(DØ$,KØ$)    ! Find which key was pressed function
ØØ2Ø DISP DØ$
ØØ3Ø 'KEY': K$=KEY$
ØØ4Ø IF NOT POS(KØ$,K$) THEN "KEY"
ØØ5Ø FNK$=K$
ØØ6Ø END DEF

**************   INITIALIZATION *************

ØØ7Ø OPTION BASE Ø
ØØ8Ø DESTROY A,H,P,T$,K$,Kl$
ØØ9Ø DIM K$[4],Kl$[l],T$[l]
Ø1ØØ INTEGER N,P

**** BEGIN ****

Ø11Ø 'START': Kl$=FNK$("Gauss,Simp,Trap,Quit","GSTQ")
Ø12Ø GOTO Kl$

**** SIMPSON, TRAPEZOIDAL START ****

Ø13Ø 'S': 'T': T$=FNK$("Explicit?(Y/N)","YN")
Ø14Ø IF T$="Y" THEN "PRT"

**** For implicit case, get function values

Ø15Ø 'PL': INPUT "Nmbr of pnts= ",STR$(N);N
Ø16Ø IF N<=Ø THEN DISP "Must be positive" @ GOTO 'PL'
Ø17Ø IF Kl$="S" AND MOD(N-l,2) THEN DISP "MUST BE ODD NUMBER" @ GOTO
"PL"
Ø18Ø INPUT "Interval Length= ",STR$(H);H
Ø19Ø DIM A(N-l)
Ø2ØØ CALL EDITl(A,N-l)
Ø21Ø GOTO "CALC"

**** For explicit case, get number of partitions ****

Ø22Ø 'PRT': INPUT "# of Partitions= ",STR$(P);P
Ø23Ø IF P<=Ø THEN DISP "Must be positive" @ GOTO 'PRT'
Ø24Ø IF Kl$="S" AND MOD(P,2) THEN DISP "MUST BE EVEN, NONZERO" @ GOTO
"PRT"

**** For explicit case, get function ****

Ø25Ø 'G': INPUT "f(x)= ",F$;F$
Ø26Ø IF Kl$="G" THEN T$="N"
Ø27Ø IF F$="" THEN "G"

**** Get the bounds for the integral ****

Ø28Ø 'B': INPUT "Lwr,Upr bounds= ",STR$(L)&","&STR$(U);L,U
Ø29Ø IF L=U THEN DISP "THE INTEGRAL VALUE IS Ø" @ GOTO "B"
```

```
0300 'CALC': O=FLAG(-16,O) @ T=FLAG(-10,1) ! Temporarily set radians
0310 DISP "calculating.."
0320 GOSUB K1$&T$
0330 O=FLAG(-16,O) @ T=FLAG(-10,T) ! Re-establish angular mode
0340 D1$=PEEK$("2F946",4) @ DELAY INF,INF ! Save wait period
0350 DISP "Result=";R ! Display result
0360 POKE "2F946",D1$ ! Re-establish wait period
0370 GOTO "START"

****** SUB PROGRAM CALL SECTION ******

0380 'GN': CALL GAUSS(F$,L,U,R) @ RETURN
0390 'SN': CALL SIMPSON(A,N-1,H,R) @ RETURN
0400 'TN': CALL TRAP(A,N-1,H,R) @ RETURN
0410 'SY': CALL SIMPSONE(F$,L,U,(P),R) @ RETURN
0420 'TY': CALL TRAPE(F$,L,U,(P),R) @ RETURN
0430 'Q': PUT "#38" @ END  ! Restore cursor and end program
0440 !

***** Trapezoidal rule, explicit case ****

0450 SUB TRAPE(F$,L,U,P,R)
0460 H=(U-L)/P
0470 Q=0
0480 X=L @ R=VAL(F$)
0490 X=U @ R=R+VAL(F$)
0500 FOR I=1 TO P-1
0510 X=L+H*I ! what is A?
0520 Q=Q+VAL(F$)
0530 NEXT I
0540 R=(R+2*Q)*H/2
0550 END SUB
0560 !

**** Trapezoidal rule, implicit case ****

0570 SUB TRAP(A(),N,H,R)
0580 R=A(0)+A(N)
0590 Q=0
0600 FOR I=1 TO N-1
0610 Q=Q+A(I)
0620 NEXT I
0630 R=(R+2*Q)*H/2
0640 END SUB
0650 !

**** Simpson's rule, implicit case ****

0660 SUB SIMPSON(A(),N,H,R)
0670 R=A(0)+A(N)
0680 FOR I=1 TO N-3 STEP 2
0690 R=R+4*A(I)+2*A(I+1)
0700 NEXT I
0710 R=(R+4*A(N-1))*H/3
```

```
0720 END SUB
0730 !

**** Simpson's rule, explicit case ****

0740 SUB SIMPSONE(F$,L,U,P,R)
0750 H=(U-L)/P
0760 X=L @ R=VAL(F$)
0770 X=U @ R=R+VAL(F$)
0780 FOR I=1 TO P-3 STEP 2
0790 X=L+H*I @ R=R+4*VAL(F$)
0800 X=X+H @ R=R+2*VAL(F$)
0810 NEXT I
0820 X=U-H
0830 R=(R+4*VAL(F$))*H/3
0840 END SUB
0850 !

**** 16 point Gaussian method ****

0860 SUB GAUSS(F$,L,U,R)
0870 DIM N(1,7)
0880 DATA .989400934992,.944575023073,.865631202388,.755404408355
0890 DATA .617876244403,.458016777657,.281603550779,.950125098376E-1
0900 DATA .271524594118E-1,.622535239386E-1,.951585116825E-1,.124628971256
0910 DATA .149595988817,.169156519395,.182603415045,.189450610455
0920 READ N(,)
0930 R=0
0940 C=(U-L)/2
0950 FOR I=0 TO 7
0960 X=N(0,I)*C+(L+U)/2
0970 R=N(1,I)*VAL(F$)+R
0980 X=-N(0,I)*C+(L+U)/2
0990 R=N(1,I)*VAL(F$)+R
1000 NEXT I
1010 R=R*C
1020 END SUB

**** Matrix Editor ****

1030 SUB EDIT1(A(),N)
1040 DEF FNK$(D0$,K0$)  ! Find which key was pressed function
1050 DISP D0$
1060 'KEY': K$=KEY$
1070 IF NOT POS(K0$,K$) THEN "KEY"
1080 FNK$=K$
1090 END DEF

1100 DEF FNF$(Y)    ! Fix display function. Temporarily alters,
                    ! then returns to original setting.
1110 D$=PEEK$("2F6DC",2) @ STD
1120 FNF$=STR$(Y)
1130 POKE "2F6DC",D$
1140 END DEF
```

```
1150 I=0
1160 'LOOP': K1$=FNK$("E("&FNF$(I)&")= "&STR$(A(I)),"Q #38#47#48")
1170 IF K1$=" " THEN 'M'
1180 IF K1$="Q" THEN 'Q'
1190 IF LEN(K1$)#3 THEN 'LOOP'
1200 K1$[1,1]="K"
1210 GOSUB K1$
1220 GOTO 'LOOP'
1230 'K47': I=MOD(I-1,N+1)          ! Move to previous position
1240 RETURN
1250 'K38': INPUT "Value= ";A(I)    ! Change value
1260 'K48': I=MOD(I+1,N+1)          ! Move forward
1270 RETURN

1280 'M': INPUT "Element= ";I       ! Get desired position
1290 IF I<0 OR I>N THEN DISP "EXCEEDS BOUNDS" @ WAIT 1 @ GOTO 'M'
1300 GOTO "LOOP"

1310 'Q': PUT "#38"                 ! Quit routine
1320 END SUB
```

Solution to F(x) = 0 on an Interval

This program provides two methods to find a real root of the equation f(x) = Ø. They are Newton's Method and the Pegasus Method. In addition, the program allows the user to find the value of the function for an input x.

Input for the Newton's method consists of the function to be solved, one initial guess, and as an option, the derivative of the function. If the derivative is not input, a numerical approximation is used.

Input for the Pegasus method consists of the function to be solved and two initial guesses that must bound the root. This implies that f(xØ)*f(xl)<Ø. The routine to calculate function values may be used to establish a legal interval.

When a root is found, the output will consist of the x value displayed to the setting of the computer and the value f(x), where x is the displayed root. It is possible that f(x) will not be exactly Ø. However, it will generally be within an acceptable range around zero based on the number of significant digits in the input. If the desired accuracy is not obtained, it may be possible to decrease the value used to check for acceptance (the variable E in both sub programs). In some instances, the function may have to be modified.

Newton's method converges to a root quickly in cases where it can find one. Its ability to locate a root depends on the function and the initial guess. It is not guaranteed to find a root. If the derivative is Ø or 5Ø iterations are peformed, the routine exits, displaying an appropriate message to the user. The number of iterations may be changed by altering the program.

The Pegasus Method is a modified regula falsi method with an estimated order of convergence superior to a secant method. For any legal interval, the method is guaranteed to converge.

The equations:

Newton's Method:

x<n+l> = x<n>-f(x<n>)/f'(x<n>)

The exit criteria is abs(x[n+l]-x[n]) <= epsilon
where epsilon is a small value.

f'(x) Approximation:

When the derivative is not given, the program uses the following approximation:

f'(x) <== (f(x+I/2) - f(x-I/2))/I

where I = .ØØØl(x) or .ØØØØØl if x= Ø.

Pegasus Method:

The Regula Falsi method used is:

$$x_{n+1} = x_n - f(x_n)\left[\frac{x_n - x_{n-1}}{f(x_n) - f(x_{n-1})}\right]$$

The approximations for the next iteration are chosen by:

if $f(x_{n+1})f(x_n) < 0$ then
$(x_{n-1}, f(x_{n-1})) \Leftarrow (x_n, f(x_n))$
if $f(x_{n+1})f(x_n) > 0$ then
$(x_{n-1}, f(x_{n-1})) \Leftarrow (x_{n-1}, f(x_{n-1})/(f(x_n)+f(x_{n+1})))$

References:

Dowell, M. & Jarrett, P.; "The Pegasus Method for Computing the Root of an Equation", BIT 12 (1972) pp. 503-508

Atkinson, Kendall E., "An Introduction to Numerical Analysis", Wiley and Sons

Carnahan, B., Luther, H.A., and Wilkes, J.O., "Applied Numerical Methods", Wiley and Sons, Inc.

Comments	Input	Display
1) Run program.		f(x)=
2) Key in the function using the character 'x' as the variable.	<function>	Root,F(x),Chngf,Quit
3) Press the capital letter of the desired operation.		
'R' begins the solve routine.	R	Pegasus, Newton
'F' finds function values for given x's.	F	X=
'C' allows the function to be changed.	C	f(x)=<function>
'Q' quits the program.	Q	Done
Go to the appropriate heading to continue.		
****** ROOT OPTION ******		Pegasus, Newton
R1) Choose desired method by pressing the appropriate capital letter.		
For Pegasus, press P. Go to step R2.	P	lower bound:
For Newton, press N. Go to step R8.	N	derivative=
R2) PEGASUS METHOD		lower bound:
R3a) Key in a lower bound.	<value> [ENDLINE]	upper bound:
R3b) To exit, press:	[ENDLINE]	Root,F(x),Chngf,Quit
R4a) Key in upper bound.	<value> [ENDLINE]	calculating..
R4b) To exit, press:	[ENDLINE]	Root,F(x),Chngf,Quit
R5a) The answer will be displayed.		x= <result>
R5b) If the interval does not bound a root a message will be displayed and you will return to step R4.		
R6) To see the function value at the root, press any key.	<any key>	f(x)=<value>

Comments	Input	Display
R7) To exit, press any key.	<any key>	Root,F(x),Chngf,Quit
R8) NEWTON METHOD		derivative =
R9a) If you want to use the derivative, key it in.	<derivative> [ENDLINE]	initial guess=
R9b) If you do not wish to enter the derivative:	[ENDLINE]	initial guess=
R10) Key in initial guess.	<guess> [ENDLINE]	disp convergence?
R11) If you wish to watch the convergence, press 'Y', else press any other key.	Y or <another key>	calculating..
R12) If a root is found, the result will be found. If a root is not found, an appropriate error message will be given.		x= <result>
R13) To see the function value at the root, press any key.	<any key>	f(x)=<value>
R14) To exit, press any key.	<any key>	Root,F(x),Chngf,Quit
******* FIND FUNCTION VALUES ******		X=
F1) Key in the value of x at which you wish the function evaluated at.	<value> [ENDLINE]	f(x) = <result>
F2) Press any key to continue.	<any key>	X=
F3a) Continue at step F2 for more values.		
F3b) To exit, press:	[ENDLINE]	Root,F(x),Chngf,Quit
****** CHANGE FUNCTION ******		f(x)=<crnt function>
C1) Key in desired function.	<function> [ENDLINE]	Root,F(x),Chngf,Quit
****** QUIT PROGRAM ******		Done
Q1) To get the prompt back: press any key.	<any key>	<blinking prompt>

EXAMPLE

This example demonstrates the various options of the program. It must be started at the beginning and followed to completion for the given keystrokes to work as listed. The display should be FIX 11.

Find a root for the function f(x) = ln(x) + 3x - 10.8074. Use the Pegasus method, then Newton's method. Then find a root for the function f(x) = 3x^6 - 22x^5 + 11.

Comments	Input	Display
1) Run program.		f(x) =
2) Key in function. Note that it is entered in a BASIC format.	ln(x)+3*x-10.8074 [ENDLINE]	Root,F(x),Chngf,Quit
3) Call the solve section.	R	Pegasus,Newton
4) Choose Pegasus method.	P	lower bound:
5) Guess at a bound:	5 [ENDLINE]	upper bound:
6) Guess at an upper bound.	10 [ENDLINE]	intrvl must bound root lower bound:
7) The interval did not bound the root. Use F(x) to find an interval.	[ENDLINE] F	Root,F(x),Chngf,Quit x=
8) Find f(5).	5 [ENDLINE] [ENDLINE]	f(x) = 5.8020379124 x=
9) Find f(10)	10 [ENDLINE] [ENDLINE]	21.495185093 x=
10) Try f(1)	1 [ENDLINE] [ENDLINE]	f(x)= -7.8074 x=
11) Since the function is continuous on the interval [1,5], and the values of the function at these points are of opposite sign, this interval bounds a root. Exit and move to the solve section.	[ENDLINE] R P	Root,F(x),Chngf,Quit Pegasus,Newton lower bound:
12) Key in lower bound	1 [ENDLINE]	upper bound:

EXAMPLE

Comments	Input	Display
13) Key in upper bound.	5 [ENDLINE]	calculating.. x= 3.21336Ø87Ø18
14) See how close f(3.21..) is to Ø.	[ENDLINE]	f(x) = Ø
15) Continue	[ENDLINE]	Root,F(x),Chgf,Quit
16) Solve using Newton's Method.	R N	Pegasus, Newton derivative=
17) Let routine approximate the derivative.	[ENDLINE]	initial guess=
18) Key in an initial guess.	1 [ENDLINE]	disp convergence?
19) Let's not display it.	N [ENDLINE] [ENDLINE]	calculating.. x= 3.21336Ø87Ø17 f(x)= Ø Root,F(x),Chngf,Quit
2Ø) We must now solve the second function.	C 3*x^6-22*x^5+11 R N	f(x)=LN(X)+3*X-1Ø.8Ø74 Root,F(x),Chngf,Quit Pegasus,Newton derivative=
21) Use the derivative this time. The intermediate results are displayed.	18*x^5-11Ø*x^4 [ENDLINE] 5 [ENDLINE] Y [ENDLINE] [ENDLINE]	initial guess= disp convergence? calculating.. 2.25Ø88 2.4794542177 1.93158885Ø68 . . .89346784412 x= .893467844Ø31 f(x)= Ø Root,F(x),Chngf,Quit
22) Quit the program.	Q	DONE

Note that some of the comments are not preceded by a line number.

```
ØØ1Ø ! SOLUTION TO F(X)=Ø
ØØ2Ø ! REV 1.Ø -- 1/25/84
ØØ3Ø F$=""
ØØ4Ø 'C': INPUT "f(x)=",F$;F$ ! get function
ØØ5Ø 'D': DISP "Root,F(x),Chngf,Quit" ! main prompt
ØØ6Ø A$=KEY$ @ IF A$="" THEN 6Ø
ØØ7Ø IF POS("RFQC",UPRC$(A$[1,1]))THEN GOTO UPRC$(A$[1,1]) ELSE GOTO 'D'

******* Find the Root *******

ØØ8Ø 'R': DISP "Pegasus,Newton" ! get desired method
ØØ9Ø M$=KEY$ @ IF M$="" THEN 9Ø
Ø1ØØ IF POS("PN",UPRC$(M$[1,1])) THEN GOSUB M$[1,1] ELSE GOTO 'R'
Ø11Ø IF R<>INF THEN GOTO 'DISPR'
Ø12Ø DISP "NO ROOT FOUND"
Ø13Ø GOTO 'RTN'
Ø14Ø 'DISPR': X=R ! display result
Ø15Ø DISP "x= ";X
Ø16Ø A$=KEY$ @ IF A$="" THEN 16Ø
Ø17Ø DISP "f(x)= ";VAL(F$) ! display function value
Ø18Ø 'RTN': A$=KEY$ @ IF A$="" THEN GOTO 'RTN' ELSE GOTO 'D'

******* Find values of f(x) given x *******

Ø19Ø 'F': ON ERROR GOTO 'XIT'
Ø2ØØ INPUT "X= ";X
Ø21Ø Y=VAL(F$)
Ø22Ø DISP "f(x)= ";Y
Ø23Ø A$=KEY$ @ IF A$="" THEN 23Ø ELSE 'F'
Ø24Ø GOTO 'F'
Ø25Ø 'XIT': OFF ERROR @ GOTO 'D'

******* Set up to call Pegasus method *******

Ø26Ø 'P': ON ERROR GOTO 'XIT'
Ø27Ø INPUT "lower bound:";L
Ø28Ø INPUT "upper bound: ";U
Ø29Ø X=L ! see if interval contains root
Ø3ØØ Y1=VAL(F$)
Ø31Ø X=U
Ø32Ø Y2=VAL(F$)
Ø33Ø IF Y1*Y2>Ø THEN DISP "intrvl must bound root" @ GOTO 'P'
Ø34Ø CALL PEG(F$,L,U,R)
Ø35Ø RETURN

******* Set up to call Newton method *******

Ø36Ø 'N': DESTROY N$
Ø37Ø ON ERROR GOSUB 'NEWTERR' ! catches no derivative option
Ø38Ø INPUT "derivative= ";D$
Ø39Ø INPUT "initial guess= ","1";XØ
Ø4ØØ DISP "disp convergence?"
```

```
0410 A$=KEY$ @ IF A$="" THEN 410
0420 IF UPRC$(A$)="Y" THEN D=1 ELSE D=0
0430 CALL NEWT(F$,D$,X0,D,R)
0440 RETURN
0450 'NEWTERR': OFF ERROR
0460 ON ERROR GOTO 'XIT'

******* Quit program *******

0470 'Q': DISP " DONE" @ END

******* Pegasus subprogram *******
The inputs are the function (F$), lower limit (X0), and upper limit
(X1). The result is returned through variable 'R'. The function
string must be a legal BASIC expression in variable 'x'; the limits
must bound a root.

0480 SUB PEG(F$,X0,X1,R)
0490 DISP "calculating.."
0500 E=.0000000001 ! error tolerance
0510 C=2*X1-X0
0520 X=X0 ! init y0,y1
0530 Y0=VAL(F$)
0540 X=X1
0550 Y1=VAL(F$)
0560 !
0570 'LP': X2=X1-Y1*((X1-X0)/(Y1-Y0))
0580 IF ABS(X2-C)<=E THEN R=X2 ELSE GOTO 'CT'
0590 GOTO 'ND'
0600 'CT': X=X2
0610 Y2=VAL(F$)
0620 C1=Y2*Y1
0630 IF C1<0 THEN X0=X1 @ Y0=Y1
0640 IF C1>0 THEN Y0=Y0*Y1/(Y1+Y2)
0650 X1=X2
0660 Y1=Y2
0670 C=X2
0680 GOTO 'LP'
0690 'ND': SUB END

******* Newton subprogram *******
The inputs are the function (F$), optional derivative (D$), initial
guess, and the display-convergence boolean. The result is returned
through 'R'. The functions must be in variable 'x'. If no root is
found, 'R' is set to the value 'inf'.

0700 SUB NEWT(F$,D$,X0,D,R)
0710 DISP "calculating.."
0720 E=.0000000001 ! error tolerance
0730 L=0 ! INIT LOOP COUNTER
0740 'LP': IF D$="" THEN GOSUB 'AD' ELSE X=X0 @ Y1=VAL(D$)
0750 IF Y1=0 THEN R=INF @ DISP "DERIVATIVE=0" @ GOTO 'ND'
0760 X=X0
0770 Y=VAL(F$)
```

```
0780 X1=X-Y/Y1
0790 IF ABS(X1-X)<=E THEN R=X1 @ GOTO 'ND'
0800 X0=X1
0810 L=L+1
0820 IF D THEN DISP X0
0830 IF L=50 THEN DISP "50 ITERATIONS" @ R=INF @ GOTO 'ND'
0840 GOTO 'LP'
0850 'AD': IF X0=0 THEN I=.000001 ELSE I=.001*X0 ! deriv. approx.
0860 X=X0+I/2
0870 Y1=VAL(F$)
0880 X=X0-I/2
0890 Y1=(Y1-VAL(F$))/I
0900 RETURN
0910 'ND': SUB END
```

This program allows the user to caculate the determinant of a real valued matrix and find the inverse or solve a system of equations for real or complex valued systems.

The method used is Gaussian elimination with partial pivoting. The matrix is decomposed into an LU form and the pivoting strategy is kept track of in a separate matrix. For a more in-depth discussion, please refer to the references.

The determinant is calculated from the decomposed matrix by multiplying the values in the main diagonal. It is found only for a real valued matrix.

The inverse is found by solving the system Ax(i) = I(i), where I(i) is the ith column of the identity matrix and x(i) is the ith column of A inverse. This is performed N times as i ranges from 1 to N.

If the inverse or solution to a system of equations is attempted that involves a singular matrix, the message "MATRIX IS SINGULAR" will be displayed.

Remarks:

The program will use the particular number display you specify before running the program.

If you pause the program during the listing of values, then exit the program, your delay will be set to inf.

References:

Johnston, R.L., "NUMERICAL METHODS, a Software Approach", John Wiley and Sons, 1982

Anton, Howard, "Elementary Linear Algebra", John Wiley and Sons, 1981

Atkinson, Kendal E., "An Introduction to Numerical Analysis", John Wiley and Sons, 1978

Comments	Input	Display
1) Run the program.		Real or Complex?
2) For real values:	R	
For complex values:	C	order?
3) Key in the number of rows:	<value>	A(1,1)=<value>
The matrix must be square.		or
		R(1,1)=<value>
You are now in the matrix editor. Please refer to that section, then continue with step 4.		keep a copy?
4) If you want to perform some operations and then make a few changes to the matrix, keep a copy of the original:	Y	
If no copy is desired:	N	Newmat,Edit,Calc,Quit
5) Choose desired action.		
To create a new matrix:	N	Real or Complex?
Cont. at step 1.		
To edit the current matrix:	E	A(1,1)=<value>
Refer to the editor instructions, then cont. at step 4.		or R(1,1)=<value>
To perform matrix operations:	C	Solv,Det,Inv,Main,Quit
Continue at step 6.		
To quit the program:	Q	<blinking prompt>
6) Calculation options.		
To solve system of equations:	S	B(1)=<value>
You are now in the matrix editor. Enter the values of the B vector for Ax=B. When you are done, you will see:		calculating..
Continue at solve section.		
To find the determinant:	D	calculating..
Continue at the determinant section.		
To find the inverse:	I	calculating..
Cont. at the inverse secton.		
To return to the main menu:	M	Newmat,Edit,Calc,Quit
Continue at step 5.		
To quit the program:	Q	<blinking display>

Comments	Input	Display
****** SOLVE ******		
S1) If a solution exists: These are the values of the result vector. Continue at step S2.	[ENDLINE] [ENDLINE] .	<value x(1)> <value x(2)> . .
If a solution does not exist, a message will be displayed: Return to step 6.		SINGULAR SYSTEM Newmat,Edit,Calc,Quit
S2) Once the last value has been displayed, you will see:		solve for new B?
S3) If you want to solve for a new B vector:	Y	B(1)=<value>
Enter the values according to the edit instructions. Continue at step S1.		
If you are done: Return to step 6.	N	Solv,Det,Inv,Main,Quit
****** DETERMINANT ******		
D1) The determinant will be dislayed: Press any key to continue. Go to step 6.	<any key>	det = <value> Solv,Det,Inv,Main,Quit
****** INVERSE ******		
I1) If the inverse exists: Cont. at step I2.		list by Col or Row?
If no inverse exists: Return to step 6.		SINGULAR SYSTEM Newmat,Edit,Calc,Quit
I2) To list result by columns:	C [ENDLINE] . .	<value A(1,1)> <value A(1,2)> . .
To list result by rows:	R [ENDLINE] . .	<value A(1,1)> <value A(1,2)> . .
I3) Once all values have been listed, you will see: Go to step 6.		Solv,Det,Inv,Main,Quit

****** MATRIX EDITOR ******

The matrix editor allows the user to move through a matrix and change element values. For complex valued matrices, the real and complex parts of an element are edited one at a time.

Movement through a matrix is accomplished with the arrow keys (left, right, up, and down), and element indices input. The movement wraps around when the boundary of a row or column is passed.

Comments	Input	Display
E1) To move from the current position:		A(I,J)
Move left:	<	A(I,J-1)= <value>
Move right:	>	A(I,J+1)= <value>
Move up:	<up arrow key>	A(I-1,J)= <value>
Move down:	<dwn arrow key>	A(I+1,J)= <value>
E2) To move to a desired element:	[SPC]	enter ROW,COLUMN
	<row>,<column> [ENDLINE]	A(<row>,<col>)= <val>
E3) To quit the editor:	Q	

EXAMPLES

Example 1.

Find the determinant and inverse of the matrix below. It is assumed that the display is fix 4.

6	3	-2	2	3
1	4	-3	4	2
2	3	-1	-2	9
4	3	Ø	2	1
3	5	-6	6	2

Comments	Input	Display
1) Run the program.		Real or Complex?
2) The values are real.	R	order?
3) Key in the number of rows:	5	
	[ENDLINE]	A(1,1)= Ø.ØØØØ
	[ENDLINE]	?
	6	
	[ENDLINE]	A(1,2)= Ø.ØØØØ
	[ENDLINE]	?
	3	
Enter the values of	[ENDLINE]	A(1,3)= Ø.ØØØØ
the matrix and use the	[ENDLINE]	?
editing features to adjust	-2	
any incorrect values.	[ENDLINE]	A(1,4)= Ø.ØØØØ
	[ENDLINE]	?
	2	
	[ENDLINE]	A(1,5)= Ø.ØØØØ
	[ENDLINE]	?
	3	
	.	.
	.	.
	[ENDLINE]	A(5,4)= Ø.ØØØØ
	[ENDLINE]	?
	6	
	[ENDLINE]	A(5,5)= Ø.ØØØØ
	[ENDLINE]	?
	2	
	[ENDLINE]	A(1,1)= 6.ØØØØ
	Q	keep a copy?
4) No copy will be needed.	N	Newmat,Edit,Calc,Quit
5) Choose calculation options.	C	Solv,Det,Inv,Man,Quit

6) Calculate the determinant.	D	calculating.. det = -200.0000
Press any key to continue.	[ENDLINE]	Solv,Det,Inv,Main,Quit
7) To find the inverse:	I	calculating.. list by Col. or Row?
8) Choose to list by column.	C	C(1,1)= Ø.2ØØØ
	[ENDLINE]	C(2,1)=-1.75ØØ
	[ENDLINE]	C(3,1)= Ø.7ØØØ
	[ENDLINE]	C(4,1)= 1.725Ø
	[ENDLINE]	C(5,1)= 1.ØØØØ
	[ENDLINE]	C(1,2)=-Ø.12ØØ
	[ENDLINE]	C(2,2)=-2.ØØØØ
	[ENDLINE]	C(3,2)= 1.28ØØ
	[ENDLINE]	C(4,2)= 2.54ØØ
	[ENDLINE]	C(5,2)= 1.4ØØØ
	[ENDLINE]	C(1,3)=-Ø.Ø4ØØ
	[ENDLINE]	C(2,3)= Ø.5ØØØ
	[ENDLINE]	C(3,3)=-Ø.24ØØ
	[ENDLINE]	C(4,3)=-Ø.57ØØ
	[ENDLINE]	C(5,3)=-Ø.2ØØØ
	[ENDLINE]	C(1,4)= Ø.ØØØØ
	[ENDLINE]	C(2,4)= 1.75ØØ
	[ENDLINE]	C(3,4)=-Ø.5ØØØ
	[ENDLINE]	C(4,4)=-1.625Ø
	[ENDLINE]	C(5,4)=-1.ØØØØ
	[ENDLINE]	C(1,5)=-3.3333E-12
	[ENDLINE]	C(2,5)=-1.5ØØØ
	[ENDLINE]	C(3,5)=-1.ØØØØ
	[ENDLINE]	C(4,5)=-1.75ØØ
	[ENDLINE]	C(5,5)=-1.ØØØØ
	[ENDLINE]	Solve,Det,Inv,Main,Quit
9) Now find the determinant.	D	calculating.. det = -200.0000
	[ENDLINE]	Solv,Det,Inv,Main,Quit
1Ø) Return to the main menu to perform the next problem.	M	Newmat,Edit,Calc,Quit

Problem 2.

Solve the following system of complex valued equations.

$$\begin{bmatrix} 2 + 3i & .7 - 1i \\ 4 - 1.3i & 4 + Øi \end{bmatrix} \begin{bmatrix} z1 \\ z2 \end{bmatrix} = \begin{bmatrix} 2 + 2li \\ 1 + 3i \end{bmatrix}$$

Because the matrix created in problem 1 is real valued, a new matrix must be created. Therefore the example problem must start with the new matrix option.

EXAMPLES

Comments	Input	Display
1) Start by creating a new matrix.	N	Real or Complex?
2) Choose complex.	C	order?
3) The order is 2.	2	
	[ENDLINE]	R(1,1)= Ø.ØØØØ
4) You are now in the editor.		
The 'R' indicates that	[ENDLINE]	?
the real portion of	2	
element (1,1) is being	[ENDLINE]	I(1,1)= -6.ØØØØ
prompted for.	[ENDLINE]	?
'I' indicates that the	3	
imaginary portion is	[ENDLINE]	R(1,2)= 3.ØØØØ
being prompted for.	[ENDLINE]	?
	.7	
	[ENDLINE]	I(1,2)= 2.ØØØØ
Continue entering the	.	.
values in the same manner.	.	.
Refer to the editor	.	.
instructions for editing	.	.
features.	.	.
5) When done, press:	Q	keep a copy?
6) Keep a copy this time.	Y	Newmat,Edit,Calc,Quit
7) Move to calculation options.	C	Solv,Det,Inv,Main,Quit
8) Choose solve option.	S	BR(1)= 4.ØØØØ
9) You are now in the editor.The prompt is for the real portion of element (1,1) from the b vector (Ax = b).	[ENDLINE]	?
	2	
	[ENDLINE]	BI(1)= Ø.ØØØØ
	[ENDLINE]	?
	21	
	[ENDLINE]	BR(2)= Ø.ØØØØ
Continue until all values are correctly entered.	.	.
	.	.
	.	.
1Ø) To quit editing:	Q	calculating..
		X(1)= 4.3565, 1.4203 i
You may use the arrow	[ENDLINE]	X(2)=-4.5681, Ø.7456 i
to view the entire display.	[ENDLINE]	solve for new B?

Comments	Input	Display
11) Let's solve for a new b vector.	Y	BR(1)= 1.0000
	[ENDLINE]	?
	9	
	[ENDLINE]	BI(1)= 21.0000
	[ENDLINE]	?
	-.22	
	[ENDLINE]	BR(2)= 1.4794
	[ENDLINE]	?
	-3.5	
	[ENDLINE]	BI(2)= 3.0000
	1	
	[ENDLINE]	BR(1)= 9.0000
	Q	calculating..
		X(1)= 0.4800,-2.0302 i
	[ENDLINE]	X(2)=-0.6952, 2.4362 i
	[ENDLINE]	solve for new B?
12) Let's exit.	N	Solv,Det,Inv,Main,Quit
	Q	<blinking display>

```
0010 ! rev 1.0
0020 DEF FNK$(D0$,K0$)   ! Find which key was hit
0030 DISP D0$
0040 'KEY': K$=KEY$
0050 IF NOT POS(K0$,K$) THEN "KEY"
0060 FNK$=K$
0070 END DEF

0080 'N': T=POS("RC",FNK$("Real or Complex?","RC"))
0090 INPUT "order?";N ! Get the number of rows and columns.

0100 N=T*N @ M=N ! Set dimensions based on type.
0110 DIM A(M,N),S(M)
0120 R2=0 @ C2=0

0130 'E': IF NOT C2 THEN GOTO 'E1'
0140 FOR I=1 TO M
0150 FOR J=1 TO N
0160 A(I,J)=A1(I,J)   ! Read in original matrix if copy was made
0170 NEXT J
0180 NEXT I

0190 'E1': CALL EDIT(A(,),M,N,T)
0200 R2=0    ! Indicate that the determinant has not been calculated.
0210 IF FNK$("keep a copy?","YN")="Y" THEN C2=1 ELSE C2=0
0220 IF NOT C2 THEN GOTO 'M'
0230 DIM A1(M,N)
0240 FOR I=1 TO M
0250 FOR J=1 TO N
0260 A1(I,J)=A(I,J) ! Save a copy if desired.
0270 NEXT J
0280 NEXT I

0290 'M': GOTO FNK$("Newmat,Edit,Calc,Quit","NECQ") ! main menu

0300 'C': GOTO FNK$("Solv,Det,Inv,Main,Quit","SDIMQ")

0310 'S': DIM B(M,T),X(M,1) ! solve invocation.
0320 'S1': CALL EDIT(B(,),M,T,0)
0330 DISP "calculating.."
0340 IF NOT R2 THEN CALL DECOMP(A(,),M,N,S()) @ R2=1
0350 IF S(M)=0 THEN DISP "SINGULAR SYSTEM" @ GOTO 'M'
0360 CALL SOLVE(A(,),X(,),B(,),S(),N)
0370 CALL LIST(X(,),M,1,T,"X")
0380 IF FNK$("solve for new B?","YN")="Y" THEN "S1" ELSE "C"

0390 'I': DIM C(M,N),B(M,1),X(M,1) ! Inverse calculation
0400 DISP "calculating.."
0410 IF NOT R2 THEN CALL DECOMP(A(,),M,N,S()) @ R2=1
0420 IF S(M)=0 THEN DISP "SINGULAR SYSTEM" @ GOTO 'M'
0430 FOR J=1 TO N
0440 FOR I=1 TO M
0450 B(I,1)=0
0460 NEXT I
```

```
0470 B(J,1)=1
0480 CALL SOLVE(A(,),X(,),B(,),S(),N)
0490 FOR I=1 TO M
0500 C(I,J)=X(I,1)
0510 NEXT I
0520 NEXT J
0530 CALL LIST(C(,),M,N,T,"C")
0540 GOTO 'C'

0550 'D': IF T=2 THEN DISP "NOT DONE FOR COMPLEX" @ GOTO 'C' ! det.
0560 DISP "calculating.."
0570 IF NOT R2 THEN CALL DECOMP(A(,),M,N,S()) @ R2=1
0580 D=A(1,1)
0590 FOR I=2 TO M
0600 D=D*A(I,I)
0610 NEXT I
0620 D=S(M)*D
0630 DISP "det= ";D
0640 A$=KEY$ @ IF A$="" THEN 640 ELSE "C" ! display till key hit

0650 'Q': PUT "#38" @ END ! restore blinking prompt and end.
```

MATRIX EDITOR. Allows a matrix of dimension MxN to be edited. The type (T) can be 1 or 2. 1 indicates real, 2 indicates complex. If T=Ø, then the routine assumes a vector has been passed. The value of T is then changed to the correct type indicator value.

```
0660 SUB EDIT(A(,),M,N,T)
0670 DEF FNK$(DØ$,KØ$) ! Get which key was hit
0680 DISP DØ$
0690 'KEY': K$=KEY$
0700 IF NOT POS(KØ$,K$) THEN "KEY"
0710 FNK$=K$
0720 END DEF

0730 DEF FNF$(Y)   ! temporarily change display setting for index vals.
0740 D9$=PEEK$("2F6DC",2) @ STD
0750 FNF$=STR$(Y)
0760 POKE "2F6DC",D9$
0770 END DEF

0780 DEF FND$  ! Create array elemnt prompt
0790 IF NOT (T=1 OR MOD(J,2)) THEN D$=S$ ELSE D$=R$
0800 D1$=D$&"("&FNF$((I+T-1)/T)
0810 IF D$[1,1]<>"B" THEN D1$=D1$&","&FNF$(INT((J+T-1)/T))
0820 IF T=2 AND NOT MOD(J,2) THEN V=-A(I,J) ELSE V=A(I,J)
0830 FND$=D1$&")= "&STR$(V)
0840 END DEF

0850 I=1 @ J=1 @ R=2 @ T$="BRBI"
0860 IF T=2 THEN R$="R" @ S$="I"
0870 IF T=1 THEN R$="A"
0880 IF T=Ø THEN R$=T$[1,N] @ S$=T$[3,N+2] @ T=N
```

```
     Figure out which key has been hit and branch to legal choice.
0890 'LOP': A$=FNK$(FND$,"Q #38#47#48#50#51") @
                                        IF POS("0134578#",A$) THEN "LOP"
0900 IF LEN(A$)=1 THEN GOSUB UPRC$(CHR$(NUM(A$)+33)) @ GOTO "LOP"
0910 A$[1,1]="K"
0920 GOSUB A$ @ GOTO "LOP"

0930 'K38': INPUT A(I,J) @ IF T=1 THEN GOTO 'K48' ! enter a value
0940 IF MOD(J,2) THEN A(I+1,J+1)=A(I,J) ELSE A(I+1,J-1)=A(I,J) @ A(I,
J)=-A(I,J)

0950 'K48': J=J+1 @ IF J<=N THEN RETURN ! move right
0960 J=1 @ I=I+T @ IF I>M-T+1 THEN I=1
0970 RETURN

0980 'K47': J=J-1 @ IF J THEN RETURN ! move left
0990 J=N @ I=I-T @ IF I<1 THEN I=M-T+1
1000 RETURN

1010 'K50': I=I-T @ IF I>0 THEN RETURN ! move up
1020 I=M-T+1 @ J=J-1 @ IF NOT J THEN J=N
1030 RETURN

1040 'K51': I=I+T @ IF I<=M-T+1 THEN RETURN ! move down
1050 I=1 @ J=J+1 @ IF J>N THEN J=1
1060 RETURN

1070 'A': ON ERROR GOTO 'A' ! user specified move
1080 INPUT "enter ROW,COLUMN";I,J
1090 IF I<1 OR J<1 THEN I=M+1
1100 IF T-1 THEN I=2*I-1 @ J=2*J-1
1110 IF I>M OR J>N THEN DISP "OUT OF BOUNDS" @ GOTO 'A'
1120 OFF ERROR @ RETURN

1130 'R': POP @ SUB END

 LIST MATRIX SUB PROGRAM. Allows a matrix to be listed by row
 or column. The array name is passed through parameter B$. The
 matrix is MxN, and the type (T) is 1 for real values, and 2
 for complex values. If a vector is passed, the routine will
 list by column. A vector is implied by B$ = 'X'.

1140 SUB LIST(A(,),M,N,T,B$)

1150 DEF FNF$(Y) ! Create index integer prompts
1160 D9$=PEEK$("2F6DC",2) @ STD
1170 FNF$=STR$(Y)
1180 POKE "2F6DC",D9$
1190 END DEF

1200 DEF FND$  ! Create element prompt
1210 IF B$="X" THEN FND$=FNF$((I+T-1)/T) ELSE
                          FND$=FNF$((I+T-1)/T)&","&FNF$((J+T-1)/T)
1220 END DEF
```

```
1230 D1$=PEEK$("2F946",4) @ DELAY INF,INF !  Temporarily change delay
1240 IF B$="X" THEN GOTO 'C'

1250 'P': DISP "list by Col. or Row?" ! get display choice
1260 A$=KEY$
1270 IF NOT POS("CR",UPRC$(A$[1,1])) THEN 'P'
1280 GOTO UPRC$(A$[1,1])

1290 'C': FOR J=1 TO N STEP T   ! display by column
1300 FOR I=1 TO M STEP T
1310 DISP B$&"("&FND$&")=";A(I,J);
1320 IF T=2 THEN DISP ",";A(I+1,J);"i";
1330 DISP
1340 NEXT I
1350 NEXT J
1360 I=1 @ J=1 @ GOTO 'E'

1370 'R': FOR I=1 TO M STEP T ! display by row
1380 FOR J=1 TO N STEP T
1390 DISP B$&"("&FND$&")=";A(I,J);
1400 IF T=2 THEN DISP ",";A(I+1,J);"i";
1410 DISP
1420 NEXT J
1430 NEXT I
1440 I=1 @ J=1

1450 'E': POKE "2F946",D1$ @ SUB END ! restore delay and exit
```

DECOMPOSITION OF MATRIX. Performs an LU decomposition of an MxN matrix using partial pivoting. The pivoting strategy is recorded in vector S.

```
1460 SUB DECOMP(A(,),M,N,S())
1470 S(M)=1
1480 FOR RØ=1 TO M-1
1490 PØ=RØ
1500 P1=A(RØ,RØ)

1510 FOR I=RØ+1 TO M ! choose largest absolute value for pivot
1520 IF ABS(A(I,RØ))>ABS(P1) THEN PØ=I @ P1=A(I,RØ)
1530 NEXT I
1540 IF A(PØ,RØ)=Ø THEN S(M)=Ø @ GOTO 'END' ! quit if singular
1550 S(RØ)=PØ
1560 IF PØ=RØ THEN GOTO 'C'

1570 FOR I=RØ TO N ! row exchange
1580 T=A(RØ,I)
1590 A(RØ,I)=A(PØ,I)
1600 A(PØ,I)=T
1610 NEXT I
1620 S(M)=-S(M)

1630 'C': FOR R1=RØ+1 TO M ! row r1 <--r1-mult*rØ
```

```
1640 M1=A(R1,RØ)/A(RØ,RØ) ! form multiplier
1650 A(R1,RØ)=M1 ! and save it.
1660 FOR E=RØ+1 TO N
1670 A(R1,E)=A(R1,E)-M1*A(RØ,E)
1680 NEXT E
1690 NEXT R1
1700 NEXT RØ
1710 IF A(M,N)=Ø THEN S(M)=Ø
1720 'END': SUB END
```

SOLVE ROUTINE. Takes an LU form matrix, pivot strategy vector, B vector, and calculates the X vector for the matrix equation Ax=b.

```
1730 SUB SOLVE(A(,),X(,),B(,),S(),N)
1740 M=N

1750 FOR I=1 TO M-1 ! Permute B and perform reduction
1760 T=B(S(I),1)
1770 B(S(I),1)=B(I,1)
1780 B(I,1)=T
1790 FOR J=I TO M-1
1800 B(J+1,1)=B(J+1,1)-B(I,1)*A(J+1,I)
1810 NEXT J
1820 NEXT I

1830 FOR I=N TO 1 STEP -1 ! back substitution
1840 X(I,1)=B(I,1)
1850 FOR J=I+1 TO M
1860 X(I,1)=X(I,1)-A(I,J)*X(J,1)
1870 NEXT J
1880 X(I,1)=X(I,1)/A(I,I)
1890 NEXT I
1900 SUB END
```

This program calculates a fast Fourier transform from a set of time domain points to a set of frequency domain points. The inverse fast Fourier transform, calculating the set of time domain points from a set of frequency domain points, may also be calculated.

The method used is a modification of the basic FFT algorithm. The modified algorithm takes advantage of the fact that series data is real, and uses the space normally reserved for the imaginary part of the complex sequence to calculate a double-length real transform. This is represented for two "N" length transforms as:

$$Z(n) = X(n) + iY(n) \qquad \emptyset < n < N \text{ data points}$$

The transform is:

$$Z(m) = X(m) + iY(m)$$

where

$$X(m) = \frac{Z(m) + Z(N-m)^*}{2}$$

$$Y(m) = \frac{Z(m) - Z(N-m)^*}{2i}$$

Z* is the complex conjugate of Z.

The time series F(n) is given by:

$$F(n) = X(2n) + Y(2n+1)$$

The transformation of this is:

$$F(m) = \sum_{n=\emptyset}^{N-1} X(2n)w^{mn} + \sum_{n=\emptyset}^{N-1} Y(2n+1)w^{mn}$$

$$= \sum_{p=\emptyset}^{N-1} X(p)w^{2mp} + \sum_{p=\emptyset}^{N-1} Y(p)(w^{2mp})(w^{m})$$

and:

$$F(m) = X(m) + Y(m)w\hat{}m \qquad (1)$$

$$F(N-m) = X^*(m) - [Y(m)w\hat{}m]^* \qquad (2)$$

Similarly, the inverse transform may be obtained from (1) and (2):

$$Z(m) = \frac{F(m) + F(N-m)^*}{2} + iw\hat{}(-m)\frac{F(m) - F(N-m)^*}{2}$$

$$Z(N-m) = \frac{F(m) + F(N-m)^*}{2}^* - iw\hat{}(-m)\frac{F(m) - F(N-m)^*}{2}^*$$

This is simply an interchange of Z(m) and F(m) in (1) and (2), and substitution of -w^(-m) for w^m.

The advantages gained from this adaptation of the general FFT algorithm for time series data are:

(a) A transform of twice the length can be handled with no increase in storage for input data.

(b) Since the calculation of the transform is structured as an interactive process, intermediate and final results are stored in the same locations used for input.

NOTE:

1. Since F(Ø) and F(N) are real only, F(N) can be stored in the imaginary location of F(Ø), i.e., f(1).

2. w^m = c^(-2im*pi/2N). This is half the minimum value of rotation normally used in an N-point transfer.

3. * denotes the complex conjugate.

REFERENCES

Brigham, E. O., THE FAST FOURIER TRANSFORM, Prentice-Hall, Inc. 1974.

FAST FOURIER TRANSFORM, HP-85 Math Pac, Hewlett-Packard, 1979.

Comments	Input	Display
1) Run the program.		TIME/FREQ. DATA? (T/F)
2) If time data is to be input, press T and continue with step 3. If frequency data is to be input, go to step 11.	[T]	# OF DATA POINTS?
TIME DOMAIN DATA		
3) Enter number of time domain data points to be used. Must be an integer power of 2 and >2. If it is not, the message "INPUT OUT OF RANGE" will be displayed and you will be prompted to enter the number again. If available memory is insufficient for the specified # of points, the message "NOT ENOUGH MEMORY" will be displayed and the program will again prompt for the number of points.	<N> [ENDLINE]	DATA POINT(nnn)?
4) The program will now prompt for data points 1 through N, where N is the # specified in 3).	<value>[ENDLN] <value>[ENDLN] . . . <value>[ENDLN]	DATA POINT(nnn)? . . . DATA POINT(N)?
5) If any mistakes were made in input, you now have the opportunity to correct the data. Press [Y] to make any changes, or [N] to go to step 6).		CHANGES? (Y/N)
	[Y]	DATA POINT TO CHANGE?
Enter point # to change.	<nnn> [ENDLINE]	<old value>
Original entry will be displayed for reference.	<new value> [ENDLINE]	CHANGES? (Y/N)
If no (further) changes are necessary, press [N].	[N]	TRANSFORMING...

Comments	Input	Display
6) Calculation of FFT has begun.		DC TERM =
7) Calculation ends and DC term is displayed. Press any key to continue.	<any key>	MAX FREQ. =
8) Display maximum frequency. Press any key to resume output.	<any key>	FREQ. DOMAIN OUTPUT:
9) Display complex data pairs of calculated frequency domain. First the real, and then the imaginary part of the data pair will be displayed. Press any key to display each value.	<any key> <any key> <any key> . . . <any key> <any key>	1R=<value> 1I=<value> 2R=<value> . . . NR=<value> NI=<value> >
10) For a new problem, go to step 1.		
FREQUENCY DOMAIN DATA		
11) If frequency data is to be input, press [F].	[F]	# OF COEFF. PAIRS?
12) Enter the number of coefficient pairs. This number must be one less than an integer power of 2 (1,3,7,...). If it is not, the message "INPUT OUT OF RANGE" will be displayed and you will be prompted to enter the number again. If available memory is insufficient for the number of pairs specified, the message "NOT ENOUGH MEMORY" will be displayed and the program will again ask for the number of coefficient pairs.	<N> [ENDLINE]	DC TERM=
13) Enter the DC term.	<DC term>	MAX FREQ. TERM=

Comments	Input	Display
14) Enter the maximum frequency term.	<max freq>	FREQ. DOMAIN DATA -
15) Enter the real and imaginary coefficients for each data pair as prompted.	<real coeff.> <imag coeff.> <real coeff.> . . . <real coeff.> <imag coeff.>	REAL(1)? IMAG(1)? REAL(2)? . . . REAL(N)? IMAG(N)? CHANGES? (Y/N)
16) If any mistakes were made in entering the coefficient pairs, they may be corrected by pressing [Y]. If no changes are necessary, press [N] and go to step 17).	[Y]	DATA PAIR TO CHANGE?
Enter # of coefficient pair to correct.	<n>[ENDLINE]	REAL(n)<old entry>
Original entry is displayed for reference.	<new value> [ENDLINE]	IMAG(n)<old entry>
	<new value> [ENDLINE]	CHANGES? (Y/N)
If no (further) changes are necessary, press [N].	[N]	TRANSFORMING...
17) Calculation of inverse FFT has begun.		TIME DOMAIN OUTPUT:
18) Calculation ends and data points are displayed. Press any key to move to next data point.	<any key> <any key> . . . <any key>	PT(1)=<value> PT(2)=<value> . . . PT(N)=<value> >
19) For a new problem, go to step 1).		

EXAMPLES

A) For the following set of time domain data points, calculate the Fourier transform to frequency data.

P(1)=1	P(9)=-1
P(2)=1.3Ø66	P(1Ø)=-1.3Ø66
P(3)=1.4142	P(11)=-1.4142
P(4)=1.3Ø66	P(12)=-1.3Ø66
P(5)=1	P(13)=-1
P(6)=.5412	P(14)=-.5412
P(7)=Ø	P(15)=Ø
P(8)=-.5412	P(16)=.5412

USER INSTRUCTIONS

Comments	Input	Display
1) Run the program.		TIME/FREQ. DATA?
2) Choose time data input.	[T]	# OF DATA POINTS?
3) Enter # of points.	16 [ENDLINE]	DATA POINT(1)?
4) Enter point values.	1 [ENDLN]	DATA POINT(2)?
	1.3Ø66 [ENDLN]	DATA POINT(3)?
	1.4142 [ENDLN]	DATA POINT(4)?
	1.3Ø66 [ENDLN]	DATA POINT(5)?
	1 [ENDLN]	DATA POINT(6)?
	.5412 [ENDLN]	DATA POINT(7)?
	Ø [ENDLN]	DATA POINT(8)?
	-.5412 [ENDLN]	DATA POINT(9)?
	-1 [ENDLN]	DATA POINT(1Ø)?
	-1.3Ø66 [ENDLN]	DATA POINT(11)?
	-1.4142 [ENDLN]	DATA POINT(12)?
	-1.3Ø66 [ENDLN]	DATA POINT(13)?
	-1 [ENDLN]	DATA POINT(14)?
	-.5412 [ENDLN]	DATA POINT(15)?
	Ø [ENDLN]	DATA POINT(16)?
	.5412 [ENDLN]	
5) FFT calculation begins.		TRANSFORMING...
6) Frequency domain output.		DC TERM=Ø
	<any key>	MAX FREQ. =Ø
	<any key>	FREQ DOMAIN OUTPUT:
		1R= 1.ØØØ1ØE+ØØØ
	<any key>	1I=-1.ØØØ1ØE+ØØØ
	<any key>	2R= Ø.ØØØØØØE+ØØØ
	<any key>	2I= Ø.ØØØØØØE+ØØØ
	<any key>	3R=-1.339454E-ØØ6
	<any key>	3I=-1.339454E-ØØ6

FREQ DOMAIN OUTPUT:

USER INSTRUCTIONS

Comments	Input	Display
	<any key>	4R= Ø.ØØØØØØE+ØØØ
	<any key>	4I= Ø.ØØØØØØE+ØØØ
	<any key>	5R= 6.134477E-ØØ6
	<any key>	5I=-6.134477E-ØØ6
	<any key>	6R= Ø.ØØØØØØE+ØØØ
	<any key>	6I= Ø.ØØØØØØE+ØØØ
	<any key>	7R=-1.5Ø2234E-ØØ5
	<any key>	7I=-1.5Ø2234E-ØØ5
Done.	<any key>	>

B) For the following set of frequency domain data pairs, perform the inverse Fourier transform to calculate the set of time domain data points. The DC term and the maximum frequency term are Ø.

REAL(1)=1	IMAG(1)=-1
REAL(2)=Ø	IMAG(2)=Ø
REAL(3)=-1.3395E-6	IMAG(3)=-1.3395E-6
REAL(4)=Ø	IMAG(4)=Ø
REAL(5)=6.1345E-6	IMAG(5)=-6.1345E-6
REAL(6)=Ø	IMAG(6)=Ø
REAL(7)=-1.5Ø22E-5	IMAG(7)=-1.5Ø22E-5

USER INSTRUCTIONS

Comments	Input	Display
1) Run the program.		TIME/FREQ. DATA? (T/F)
2) Enter frequency data.	[F]	# OF COEFF. PAIRS?
3) Seven frequency data pairs.	7 [ENDLINE]	DC TERM=
4) DC term is Ø.	Ø [ENDLINE]	MAX FREQ. TERM=
5) Maximum frequency term is Ø.	Ø [ENDLINE]	FREQ. DOMAIN DATA -
6) Begin entry of frequency domain data pairs.		REAL(1)?
	1 [ENDLINE]	IMAG(1)?
	-1 [ENDLINE]	REAL(2)?
	Ø [ENDLINE]	IMAG(2)?
	Ø [ENDLINE]	REAL(3)?
	-1.3395E-6 [ENDLINE]	IMAG(3)?
	-1.3395E-6 [ENDLINE]	REAL(4)?
	Ø [ENDLINE]	IMAG(4)?

Comments	Input	Display
	Ø [ENDLINE]	REAL(5)?
	6.1345E-6 [ENDLINE]	IMAG(5)?
	-6.1345E-6 [ENDLINE]	REAL(6)?
	Ø [ENDLINE]	IMAG(6)?
	Ø [ENDLINE]	REAL(7)?
	-1.5Ø22E-5 [ENDLINE]	IMAG(7)?
	-1.5Ø22E-5 [ENDLINE]	
7) Inverse FFT calculation begins.		TRANSFORMING...
8) Display time domain data points.		TIME DOMAIN OUTPUT:
		PT(1)= 9.999898E-ØØ1
	<any key>	PT(2)= 1.3Ø6587E+ØØØ
	<any key>	PT(3)= 1.414186E+ØØØ
	<any key>	PT(4)= 1.3Ø6587E+ØØØ
	<any key>	PT(5)= 9.999898E-ØØ1
	<any key>	PT(6)= 5.411945E-ØØ1
	<any key>	PT(7)= 1.ØØØØØØE-Ø12
	<any key>	PT(8)=-5.411945E-ØØ1
	<any key>	PT(9)=-9.999898E-ØØ1
	<any key>	PT(1Ø)=-1.3Ø6587E+ØØØ
	<any key>	PT(11)=-1.414186E+ØØØ
	<any key>	PT(12)=-1.3Ø6587E+ØØØ
	<any key>	PT(13)=-9.999898E-ØØ1
	<any key>	PT(14)=-5.411945E-ØØ1
	<any key>	PT(15)=-1.ØØØØØØE-Ø12
	<any key>	PT(16)= 5.411945E-ØØ1
Done.	<any key>	>

```
Ø01Ø ! FOURIER TRANSFORM
Ø02Ø ! Revision 1.Ø  4/16/84
Ø03Ø !
Ø04Ø F9=FLAG(5,FLAG(-1Ø))
Ø05Ø OPTION BASE 1 @ OPTION ANGLE RADIANS @ STD @ DELAY Ø,Ø
Ø06Ø ON ERROR GOTO 'ERR'
Ø07Ø I1$="TF" @ I2$="YN"
Ø08Ø O1$="3D,'R=',MZ.6DE" @ O2$="3D,'I=',MZ.6DE"
Ø09Ø O3$="'PT(',3D,')=',MZ.6DE"
Ø1ØØ DISP 'TIME/FREQ. DATA? (T/F)'
Ø11Ø 'W1': K1$=UPRC$(KEY$) @ IF NOT POS(I1$,K1$) THEN 'W1'
Ø12Ø IF K1$='F' THEN SFLAG 1 @ F=-1 ELSE CFLAG 1 @ F=1

****  Input data points for FFT, or coefficient pairs for  ****
****  inverse FFT. Number of data points must be a power   ****
****  of 2 and greater than 2. Number of coefficient       ****
****  pairs must be one less than a power of 2.            ****

Ø13Ø 'IN':
Ø14Ø IF NOT FLAG(1) THEN INPUT '# OF DATA POINTS?';N @ GOTO 'INA'
Ø15Ø INPUT '# OF COEFF. PAIRS?';N
Ø16Ø N=N*2+2
Ø17Ø 'INA': P=1
Ø18Ø IF N=2 THEN 'OR'
Ø19Ø FOR L=1 TO 1Ø
Ø2ØØ P=P*2
Ø21Ø IF P=N THEN P1=L @ N2=N/2 @ GOTO 'START'
Ø22Ø NEXT L
Ø23Ø 'OR':
Ø24Ø DISP 'INPUT OUT OF RANGE' @ WAIT 1 @ GOTO 'IN'
Ø25Ø 'START':
Ø26Ø DIM R(N2),I(N2)
Ø27Ø IF FLAG(1) THEN 'IFFT'

****  Input time domain data for FFT  ****

Ø28Ø 'FFT':
Ø29Ø J=Ø
Ø3ØØ FOR L=1 TO N2
Ø31Ø J=J+1 @ DISP 'DATA POINT(';J;')'; @ INPUT R(L)
Ø32Ø J=J+1 @ DISP 'DATA POINT(';J;')'; @ INPUT I(L)
Ø33Ø NEXT L

****  Changes to FFT data  ****

Ø34Ø 'C1':
Ø35Ø DISP 'CHANGES? (Y/N)'
Ø36Ø 'W2': K2$=UPRC$(KEY$) @ IF NOT POS(I2$,K2$) THEN 'W2'
Ø37Ø IF K2$="N" THEN DISP 'TRANSFORMING...' @ GOTO 'FFTC'
Ø38Ø 'C2': INPUT 'DATA POINT TO CHANGE?';L
Ø39Ø IF L<=Ø OR L>2*N2 THEN 'C2'
Ø4ØØ I2=L/2
Ø41Ø IF I2=INT(I2) THEN J=I2 ELSE J=INT(I2)+1
Ø42Ø DISP 'DATA POINT(';L;')';
```

```
0430 IF J=I2 THEN INPUT '',STR$(I(J));I(J) ELSE INPUT '',STR$(R(J));R(J)
0440 GOTO 'C1'

****  Input DC term, maximum frequency and    ****
****  frequency domain data for inverse FFT.  ****

0450 'IFFT':
0460 INPUT 'DC TERM=';R(1)
0470 INPUT 'MAX FREQ. TERM=';I(1)
0480 DISP 'FREQ. DOMAIN DATA -' @ WAIT 1
0490 FOR L=2 TO N2
0500 DISP 'REAL(';L-1;')'; @ INPUT R(L)
0510 DISP 'IMAG(';L-1;')'; @ INPUT I(L)
0520 NEXT L

****  Changes to inverse FFT data ****

0530 'C3':
0540 DISP 'CHANGES? (Y/N)'
0550 'W3': K3$=UPRC$(KEY$) @ IF NOT POS(I2$,K3$) THEN 'W3'
0560 IF K3$='N' THEN DISP 'TRANSFORMING...' @ GOTO 'IFFTC'
0570 'C4': INPUT 'COEFF. PAIR TO CHANGE?';L
0580 IF L<=0 OR L>N2-1 THEN 'C4'
0590 DISP 'REAL(';L;')'; @ INPUT '',STR$(R(L+1));R(L+1)
0600 DISP 'IMAG(';L;')'; @ INPUT '',STR$(I(L+1));I(L+1)
0610 GOTO 'C3'

****  Start FFT calculation  ****

0620 'FFTC':
0630 K=0
0640 FOR J=1 TO N2-1
0650 L=2
0660 IF K<N2/L THEN 680
0670 K=K-N2/L @ L=L+L @ GOTO 660
0680 K=K+N2/L
0690 IF K<=J THEN 710
0700 A=R(J+1) @ R(J+1)=R(K+1) @ R(K+1)=A @ A=I(J+1) @ I(J+1)=I(K+1)
@ I(K+1)=A
0710 NEXT J
0720 G=.5 @ P2=1
0730 FOR L=1 TO P1-1
0740 G=G+G @ C=1 @ E=0 @ Q=SQR((1-P2)/2)*F
0750 P2=(1-2*(L=1))*SQR((1+P2)/2)
0760 FOR M=1 TO G
0770 FOR J=M TO N2 STEP G+G
0780 K=J+G @ A=C*R(K)+E*I(K) @ B=E*R(K)-C*I(K)
0790 R(K)=R(J)-A @ I(K)=I(J)+B @ R(J)=R(J)+A @ I(J)=I(J)-B
0800 NEXT J
0810 A=E*P2+C*Q @ C=C*P2-E*Q @ E=A
0820 NEXT M
0830 NEXT L
0840 IF FLAG(1) THEN 'IFFTO'
```

```
****  Start inverse FFT calculation  ****

0850 'IFFTC':
0860 A=PI/N2 @ P2=COS(A) @ Q=F*SIN(A)
0870 A=R(1) @ R(1)=A+I(1) @ I(1)=A-I(1)
0880 IF NOT FLAG(1) THEN R(1)=R(1)/2 @ I(1)=I(1)/2
0890 C=F @ E=0
0900 FOR J=2 TO N2/2
0910 A=E*P2+C*Q @ C=C*P2-E*Q @ E=A @ K=N2-J+2 @ A=R(J)+R(K)
0920 B=(I(J)+I(K))*C-(R(J)-R(K))*E @ U=I(J)-I(K)
0930 V=(I(J)+I(K))*E+(R(J)-R(K))*C
0940 R(J)=(A+B)/2 @ I(J)=(U-V)/2 @ R(K)=(A-B)/2 @ I(K)=-(U+V)/2
0950 NEXT J
0960 I(N2/2+1)=-I(N2/2+1)
0970 IF FLAG(1) THEN 'FFTC'
0980 FOR J=1 TO N2
0990 R(J)=R(J)/N2 @ I(J)=I(J)/N2
1000 NEXT J

****  FFT output  ****

1010 'FFTO':
1020 DISP 'DC TERM =';R(1) @ GOSUB 'WAIT'
1030 DISP 'MAX FREQ. =';I(1) @ GOSUB 'WAIT'
1040 DISP 'FREQ DOMAIN OUTPUT:' @ WAIT 1
1050 FOR L=2 TO N2
1060 DISP USING O1$;L-1,R(L) @ GOSUB 'WAIT'
1070 DISP USING O2$;L-1,I(L) @ GOSUB 'WAIT'
1080 NEXT L
1090 GOTO 'DONE'

****  Inverse FFT output  ****

1100 'IFFTO':
1110 DISP 'TIME DOMAIN OUTPUT:' @ WAIT 1
1120 J=1
1130 FOR L=1 TO N2
1140 J=J+1
1150 DISP USING O3$;J-1,R(L) @ GOSUB 'WAIT'
1160 DISP USING O3$;J,I(L) @ GOSUB 'WAIT'
1170 J=J+1
1180 NEXT L
1190 'DONE': F9=FLAG(-10,FLAG(5)) @ PUT '#43' @ END
1200 'WAIT': IF KEY$='' THEN 'WAIT' ELSE RETURN
1210 'ERR': IF ERRL=260 THEN DISP 'NOT ENOUGH MEMORY' @ GOTO 'IN'
1220 DISP ERRM$ @ GOTO 'DONE'
```

Polynomial Root Finder

This program finds all solutions, both real and complex, of P(x)=Ø, where P is a polynomial of the form:

$$P(x)=a(n)x^n + a(n-)x^(n-1) + \ldots + a(1)x + a(Ø) = Ø$$

Inputs to the program are the degree of the polynomial, the real coeficients a(n)...a(Ø), tolerances for the evaluation of the function and for each root, and the maximum number of iterations per root.

This program uses Laguerre's method to find the roots of the specified polynomial by computing a sequence of approximations Z(1), Z(2),..., to a root using the formula Z(k+1)=Z(k)+S(k). S(k) is called the Laguerre step, and is defined as:

$$\frac{-nP(Z(k))}{P'(Z(k))+[(n-1)^2(P'(Z(k))^2-n(n-1)P''(z(k))]^.5}$$

where P, P', and P'' are the value of the polynomial and its first and second derivatives evaluated at the current iterate k, and n is the degree of the polynomial. The sign in the denominator is chosen to give the Laguerre step of smaller size, which in most cases insures that the roots will be found in order of increasing magnitude.

After an iterate is accepted as a root, synthetic division is used to deflate the polynomial by the factor (x-r) if the root is real, or (x^2-2Re(r)+!!r!!^2) if the root is complex. This saves arithmetic operations, and prevents repetitive convergence to the same root.

For polynomials with only real roots, Laguerre's method will always converge to a root for any choice of real initial estimate. However, for roots of high multiplicity, some loss of accuracy may be observed. If complex roots are present, this method will usually converge to a valid root. If it does not, provisions are made for supplying a new initial estimate and starting the process again.

REFERENCES

Dahlquist, G. and Bjorck, A. NUMERICAL METHODS. Prentice-Hall, 1974.

HP-75 MATH PAC. Hewlett-Packard, 1983

COMMENTS	INPUT	DISPLAY
1) Run the program.		POLYNOMIAL ROOT FINDER ORDER OF POLYNOMIAL?
2) Input degree of polynomial. Must be a positive integer greater than 1. If it is not, you will be asked to enter it again.	<n>	A(n)=?
3) Enter coefficients of each term, starting with the highest-ordered term.	<a(n)> [ENDLINE]	A(n-1)=?
	<a(n-1)> [ENDLINE]	.
	.	.
	.	.
	.	A(Ø)=?
	<a(Ø)> [ENDLINE]	TOL. FOR ROOTS=1.E-1Ø
4) Input tolerance for roots. Default value of 1E-1Ø is displayed. If the magnitude of the Laguerre step is less that this value (and 5) is also satisfied) then the current iterate is accepted as a root.	<new value> [ENDLINE]	TOL. FOR FCN=1.E-8
5) Input tolerance for evaluation of function. default value of 1E-8 is displayed. If P(x) for the current iterate is less than this value, and step 4) has been satisfied, the current iterate is accepted as a root.	<new value> [ENDLINE]	MAX # OF ITERATIONS=
6) Input maximum number of iterations for each root. If this number is exceeded before a valid root is found, the message 'NO CONVERGENCE' will be displayed and the user will be allowed to specify a new initial iterate and start the search again.	<nnn> [ENDLINE]	LOOKING FOR ROOTS...

COMMENTS	INPUT	DISPLAY
7) Calculation of roots has begun. As each root is found, the program will indicate how many roots have been found to this point.		# OF ROOTS FOUND = 1 # OF ROOTS FOUND = 2 . . . # OF ROOTS FOUND = n
8) The real and imaginary parts of each root are displayed. Pressing any key will continue the displaying of the roots.		ROOT# 1: R=<value>
	<any key>	ROOT# 1: I=<value>
	<any key>	ROOT# 2: R=<value>
	<any key>	.
	.	.
	.	.
	.	ROOT# N: R=<value>
	<any key>	ROOT# N: I=<value>
	<any key>	>
9) Done.		

EXAMPLES

A) Find the roots of the polynomial given below. Use default values for the tolerances and limit iterations to 10.

$$P(x) = 5x^6-45x^5+225x^4-425x^3+170x^2+370x-500$$

USER INSTRUCTIONS

COMMENTS	INPUT	DISPLAY
1) Run the program.		POLYNOMIAL ROOT FINDER ORDER OF POLYNOMIAL?
2) Enter order of polynomial.	6 [ENDLINE]	A(6)=?
3) Enter the coefficients, starting with the highest order term.	5 [ENDLINE]	A(5)=?
	-45 [ENDLINE]	A(4)=?
	225 [ENDLINE]	A(3)=?
	-425 [ENDLINE]	A(2)=?
	170 [ENDLINE]	A(1)=?
	370 [ENDLINE]	A(0)=?
	-500 [ENDLINE]	TOL. FOR ROOTS=1.E-10
4) Use default value.	[ENDLINE]	TOL. FOR FCN=1.E-8
5) Use default value.	[ENDLINE]	MAX # OF ITERATIONS=
6) Limit to 10 iterations.	10 [ENDLINE]	LOOKING FOR ROOTS...
7) Calculation of roots begins.		# OF ROOTS FOUND = 2 # OF ROOTS FOUND = 3 # OF ROOTS FOUND = 4 # OF ROOTS FOUND = 6
8) Display real and imaginary parts of each root.		ROOT# 1: R= 1.000000E+000
	<any key>	ROOT# 1: I= 1.000000E+000
	<any key>	ROOT# 2: R= 1.000000E+000
	<any key>	ROOT# 2: I=-1.000000E+000
	<any key>	ROOT# 3: R=-1.000000E+000
	<any key>	ROOT# 3: I= 0.000000E+000
	<any key>	ROOT# 4: R= 2.000000E+000
	<any key>	ROOT# 4: I= 0.000000E+000
	<any key>	ROOT# 5: R= 3.000000E+000
	<any key>	ROOT# 5: I=-4.000000E+000
	<any key>	ROOT# 6: R= 3.000000E+000
	<any key>	ROOT# 6: I= 4.000000E+000
	<any key>	>
9) Done.		

```
0010 DEFAULT ON
0020 ! POLYNOMIAL ROOT FINDER
0030 ! Revision 1.00  4/8/84
0040 STD @ OPTION BASE 0 @ INTEGER I,J,K,N
0050 A0$='YN' @ A7$='R' @ A8$='I'
0060 A9$="'ROOT#',DD,': ',A,'=',MZ.6DE"
0070 DEF FNF2(U1,V1)=SQRT(U1*U1+V1*V1)
0080 DISP "POLYNOMIAL ROOT FINDER" @ WAIT .5

****  INPUT ORDER, COEFFICIENTS, TOLERANCES  ****
****  AND MAXIMUM NUMBER OF ITERATIONS       ****

0090 'ORD':
0100 INPUT "ORDER OF POLYNOMIAL? ";N
0110 IF N<=1 OR N#IP(N) THEN DISP "INVALID ORDER" @ GOTO 'ORD'
0120 ON ERROR GOTO 'MEM'
0130 DIM A0(N),R0(N,2),L(2),C9(N)
0140 FOR I=N TO 0 STEP -1
0150 DISP 'A(';STR$(I);') ='; @ INPUT A0(I)
0160 NEXT I
0170 'TOL':
0180 E1=.0000000001 @ SCI 0
0190 INPUT 'TOL. FOR ROOTS=',STR$(E1);E1
0200 INPUT 'TOL. FOR FCN=',STR$(E1*100);E2
0210 STD
0220 'IT': INPUT 'MAX # OF ITERATIONS=';I0
0230 IF IP(I0)#I0 OR NOT I0 THEN 'IT'
0240 DISP 'LOOKING FOR ROOTS...' @ WAIT .5
0250 J=N @ K=0 @ X=0 @ Y=0
0260 'FINDR':

****  IS ZERO A ROOT?  ****

0270 IF NOT A0(0) THEN 'FAR'

****  INPUT NEW GUESS IF ITERATION LIMIT EXCEEDED  ****

0280 'LOOP': K=K+1
0290 IF K>I0 THEN DISP 'NO CONVERGENCE' @ WAIT 1 @ GOSUB 'NUROOT'

****  CALCULATE P, P', AND P'' AT Z(x,y)  ****

0300 R=X*X+Y*Y @ D=X+X
0310 D0=0 @ D1=0 @ C0=0 @ C1=0
0320 B1=A0(J) @ B0=A0(J-1)+D*A0(J)
0330 FOR I=J-2 TO 0 STEP -1
0340 IF I=0 THEN D=X
0350 IF I>J-4 THEN 380
0360 V=D1*R @ D1=D0
0370 D0=C1+D*D0-V
0380 V=C1*R @ C1=C0
0390 C0=B1+D*C0-V
0400 V=B1*R @ B1=B0
0410 B0=A0(I)+D*B0-V
0420 NEXT I
```

```
****  BEGIN CALCULATION OF LAGUERRE STEP  ****

Ø43Ø P(1,1)=BØ @ P(1,2)=B1*Y
Ø44Ø P(2,1)=B1-2*Y*Y*C1 @ P(2,2)=2*Y*CØ
Ø45Ø P(3,1)=2*CØ-8*Y*Y*DØ @ P(3,2)=2*Y*(3*C1-4*Y*Y*D1)
Ø46Ø CALL MULT(P(1,1),P(1,2),P(3,1),P(3,2),S1,S2)
Ø47Ø S1=-S1*J*(J-1) @ S2=-S2*J*(J-1)
Ø48Ø CALL MULT(P(2,1),P(2,2),P(2,1),P(2,2),S3,S4)
Ø49Ø S3=S3*(J-1)^2 @ S4=S4*(J-1)^2
Ø5ØØ CALL ADD(S3,S4,S1,S2,S1,S2)
Ø51Ø CALL SQAROOT(S1,S2,S1,S2)
Ø52Ø CALL ADD(P(2,1),P(2,2),S1,S2,L1,L2)
Ø53Ø CALL ADD(P(2,1),P(2,2),-S1,-S2,L3,L4)
Ø54Ø CALL DIVID(P(1,1),P(1,2),L1,L2,L1,L2)
Ø55Ø CALL DIVID(P(1,1),P(1,2),L3,L4,L3,L4)

****  CHOOSE SIGN OF DENOMINATOR TO  ****
****  PRODUCE SMALLER LAGUERRE STEP  ****

Ø56Ø IF FNF2(L1,L2)<FNF2(L3,L4) THEN S1=L1 @ S2=L2 ELSE S1=L3 @ S2=L4
Ø57Ø L(1)=-J*S1 @ L(2)=-J*S2

****  FORCE REAL ROOT  ****

Ø58Ø IF ABS(L(2))<.ØØØ1*FNF2(L(1),L(2)) THEN L(2)=Ø @ Y=Ø
Ø59Ø X=X+L(1) @ Y=Y+L(2)

****  CHECK FOR VALID ROOT  ****

Ø6ØØ IF FNF2(L(1),L(2))>ABS(E1) THEN 'LOOP'
Ø61Ø IF FNF2(P(1,1),P(1,2))<ABS(E2) THEN 'FAR'
Ø62Ø DISP 'INVALID ROOT FOUND' @ WAIT 1 @ GOSUB 'NUROOT' @ GOTO 'LOOP'

****  FOUND VALID ROOT(S) - IF COMPLEX  ****
****  ROOT ASSUME CONJUGATE IS A ROOT   ****

Ø63Ø 'FAR':
Ø64Ø RØ(J,1)=X @ RØ(J,2)=Y @ IF Y THEN GOSUB 'ROOTI'
Ø65Ø DISP '# OF ROOTS FOUND =';N-J+1
Ø66Ø IF NOT FNF2(RØ(J,1),RØ(J,2)) THEN GOSUB 'ROOTØ' ELSE GOSUB 'DEFLATE'
Ø67Ø J=J-1 @ K=Ø @ IF NOT J THEN 'DR'
Ø68Ø IF J=1 THEN RØ(1,1)=-AØ(Ø)/AØ(1) ELSE K=Ø @ X=Ø @ Y=Ø @ GOTO 'FINDR'

****  DISPLAY ROOTS  ****

Ø69Ø DISP '# OF ROOTS FOUND =';N
Ø7ØØ 'DR':
Ø71Ø FOR I=N TO 1 STEP -1
Ø72Ø DISP USING A9$;N-I+1,A7$,RØ(I,1) @ GOSUB 'WAIT'
Ø73Ø DISP USING A9$ N-I+1,A8$,RØ(I,2) @ GOSUB 'WAIT'
Ø74Ø NEXT I
Ø75Ø 'DONE': PUT "#43" @ END
Ø76Ø 'WAIT': IF KEY$='' THEN 'WAIT' ELSE RETURN
```

```
****  DEFLATION ROUTINES - IF ROOT IS REAL, DEFLATE BY  ****
****  LINEAR FACTOR (x-r). IF ROOT IS COMPLEX, DEFLATE  ****
****  BY BINOMIAL X^2-2Re(r)X+!!r!!^2.                  ****

0770 'DEFLATE':
0780 IF Y THEN 'DEFLATEI'
0790 C9(J-1)=A0(J)
0800 FOR I=J-1 TO 1 STEP -1
0810 C9(I-1)=A0(I)+C9(I)*R0(J,1)
0820 NEXT I
0830 FOR I=0 TO J-1 @ A0(I)=C9(I) @ NEXT I
0840 RETURN
0850 'DEFLATEI':
0860 C9(J-1)=A0(J+1) @ C9(J)=0 @ IF J=1 THEN RETURN
0870 FOR I=J-1 TO 1 STEP -1
0880 C9(I-1)=A0(I+1)+C9(I)*R0(J,1)*2-(R0(J,1)^2+R0(J,2)^2)*C9(I+1)
0890 NEXT I
0900 FOR I=0 TO J-1 @ A0(I)=C9(I) @ NEXT I
0910 RETURN

****  DEFLATE FOR ZERO ROOT  ****

0920 'ROOT0':
0930 FOR I=0 TO J-1 @ A0(I)=A0(I+1) @ NEXT I
0940 RETURN

****  SET NEXT ROOT TO COMPLEX CONJUGATE  ****
****  OF COMPLEX ROOT JUST FOUND          ****

0950 'ROOTI':
0960 J=J-1 @ R0(J,1)=X @ R0(J,2)=-Y
0970 RETURN

****  ASK FOR NEW GUESS IF FAILURE TO CONVERGE  ****

0980 'NUROOT':
0990 DISP 'NEW GUESS? (Y/N)'
1000 'W0': A1$=KEY$ @ IF NOT POS(A0$,UPRC$(A1$)) THEN 'W0'
1010 IF UPRC$(A1$)='N' THEN 'DONE'
1020 INPUT 'NEW u =',STR$(X);X
1030 INPUT 'NEW v =',STR$(Y);Y
1040 K=0
1050 RETURN
1060 'MEM':
1070 IF ERRN#24 THEN DISP ERRM$ @ GOTO 'DONE'
1080 DISP 'LOW MEM - REDUCE ORDER'  @ WAIT 1 @ GOTO 'ORD'

****  ADDITION OF COMPLEX NUMBERS  ****

1090 SUB ADD(U1,V1,U2,V2,U,V)
1100 U=U1+U2 @ V=V1+V2
1110 END SUB
```

```
****  MULTIPLICATION OF COMPLEX NUMBERS  ****

1120 SUB MULT(U1,V1,U2,V2,U,V)
1130 U=U1*U2-V1*V2 @ V=U1*V2+U2*V1
1140 END SUB

****  DIVISION OF COMPLEX NUMBERS  ****

1150 SUB DIVID(U1,V1,U2,V2,U,V)
1160 CALL MULT(U1,V1,U2,-V2,Z1,Z2)
1170 D5=U2*U2+V2*V2 @ IF NOT D5 THEN U=0 @ V=0 @ GOTO 1200
1180 U=Z1/D5
1190 V=Z2/D5
1200 END SUB

****  SQUARE ROOT OF A COMPLEX NUMBER  ****

1210 SUB SQAROOT(U1,V1,U,V)
1220 A2=SQR((SQR(U1*U1+V1*V1)+ABS(U1))/2)
1230 IF NOT A2 THEN U=0 @ V=0 @ GOTO 1280
1240 B2=V1/(2*A2)
1250 IF U1>=0 THEN U=A2 @ V=B2 @ GOTO 1280
1260 U=ABS(B2)
1270 IF B2>=0 THEN V=A2 ELSE V=-A2
1280 END SUB
```

NOTES

NOTES

NOTES

00071-90064

Printed in USA